

Formal Specification of Control Software for a Radiation Therapy Machine (Revised)

Jonathan Jacky *
Michael Patrick
Jonathan Unger

Radiation Oncology Department RC-08
University of Washington
Seattle, WA 98195

Technical Report 95-12-01

January 9, 1997

Abstract

This report presents a formal (mathematical) specification for the operator's console of a computer-controlled radiation therapy machine equipped with a multileaf collimator. This formal specification, rather than the prose specification, serves as the primary reference source for programming and test planning.

Specified functions include selecting treatment setups from a database of stored prescriptions, setting up prescriptions on the treatment machine manually or semi-automatically, checking that the setup conforms to the prescription (with provision for overriding certain settings, with operator confirmation), safety interlocking and essential user interface features. The specification supports physics and experimental procedures as well as normal patient treatments.

The specification is expressed in the Z notation. It formalizes the requirements in a thorough informal (English prose) specification. Its organization suggests a detailed design.

*email `jon@radonc.washington.edu`, telephone (206)-548-4117, fax (206)-548-6218

©1994,1995,1996,1997 by Jonathan Jacky, Michael Patrick and Jonathan Unger

This work may not be copied or reproduced in whole or part for any commercial purpose. Permission to photocopy in whole or part without payment of fee is granted for nonprofit educational and research purposes provided that all such copies include the following notice: a notice that such copying is by permission of the authors; an acknowledgment of the authors of the work; and all applicable portions of this copyright notice. All rights reserved.

1 Introduction

This report presents portions of a formal specification for a real medical device, a radiation therapy machine¹. This specification, rather than the informal prose description, serves as the primary reference source for programming and test planning. A paper [4] describes the development of part of the program based on the contents of this report.

This formal specification is based on a thorough informal (English prose) specification presented as Chapters 2 and 8 in [2]. Here we attempt to formalize the requirements in that source. We have included many cross-references. Decimal numbers and integers (as in 8.4, 191) refer to chapter, section and page numbers in [2], respectively.

The formal specification is expressed in the Z notation [7]. We have corrected syntax and type errors detected by a checker [6].

2 Overview

Much of the apparent complexity in the prose requirements arises from the interaction of several subsystems which, by themselves, are simpler. In the formal specification we partition the system into subsystems and describe simple operations on each. For each operation on the system as a whole, we define a separate operation on each affected subsystem. The complex behaviors of the whole system emerge when we compose these simpler operations together.

Each subsystem is modelled by a Z state schema and a number of operation schemas on that state. This partition can itself be represented in Z.

TherapyControl —

Session

Field

Intlk

...

Console

Session (section 5) models those aspects of the treatment session that are related to the prescription database (section 4 models the database itself). *Field* (section 6) models the many settings that characterize a single field. *Intlk* (section 7) models software interlocks

¹We plan to include additional portions in future versions of this report. The present version supercedes an earlier report [3] and several earlier versions of this report.

and other flags that indicate readiness. *Console* (section 8) models the user interface. Section 9 combines operations from *Session*, *Field* and *Console*.

Related operations in different subsystems are distinguished by suffix: *ExptModeS*, *ExptModeF*, and *ExptModeC* are operations in the *Session*, *Field* and *Console* subsystems, respectively.

Each user interface operation in the *Console* subsystem ensures that corresponding operations in the *Session* and *Field* subsystems are only invoked when their preconditions are satisfied. Therefore only the *Console* operations need to be total; usually there is no need to define total operations in the *Session* and *Field* subsystems. For example the *ExptModeC* operation in the *Console* subsystem checks the precondition that only physicists can invoke this operation; in *Console* we define what happens when an operator who is not a physicist attempts to enter experiment mode. Therefore *ExptModeS* and *ExptModeF* can assume that this precondition has been satisfied and need not cover the other cases.

3 System configuration

Fixed aspects of the system configuration are represented by Z global constants: sets, functions and relations. This section introduces some of these constants. It should be possible to accommodate some configuration changes simply by changing their values. All the basic types and global constants defined in this report are collected in Appendix C.

3.1 Settings and registers

The state of the therapy machine is largely determined by the values of named *items*. A glossary of items appears in Appendix A. At this writing the list of items is

```
ITEM ::= nfrac | dose_tot | dose | wedge | w_rot | filter | leaf0 | leaf39 |
gantry | collim | turnt | lat | longit | height | doseB | top |
pt_mode | pt_factor | press | temp | d_rate | t_fac |
calvolt1 | calvolt2 | p_dose | p_time | e_time
```

For brevity we omit formal declarations of the other collimator *leaves*, *leaf1 .. leaf38*.

There are many items but we can identify different subsets, where all of the members of each subset are treated the same way for some particular purpose. A glossary of item groups, and tables showing the group membership of each item, appear in Appendix B.

Settings are items which are included in field prescriptions. Other items are kept in *registers*.

In particular *dose-reg* items include calibration factors and other items concerned with the dosimetry system².

| *setting, dose-reg* : \mathbb{P} ITEM

At this writing

$\langle setting, dose_reg \rangle$ partition ITEM

$dose_reg = \{pt_mode, pt_factor, press, temp, d_rate, t_fac,$
 $calvolt1, calvolt2, p_dose, p_time, e_time\}$

Scales are items that are continuously variable over some range; examples are gantry angle and every collimator leaf position. *Selections* can only take on certain discrete values; examples are wedge and flattening filter selection. *Counters* accumulate during treatment runs; examples are dose and the number of fractions.

| *scale, selection, counter* : \mathbb{P} ITEM

At this writing

$\langle selection, scale, counter \rangle$ partition ITEM

$counter = \{nfrac, dose_tot, dose\}$
 $selection = \{wedge, w_rot, filter, pt_mode\}$

A field is prescribed by determining the values of certain of its settings. Therapy fields are defined by the values of particular settings called *prescriptions*. Experiment fields are defined by the values of settings called *presets* (8.2, 171 third bullet; Table 8.2, 173). Readiness is determined by checking all the *preset* settings in experiment mode, and all the *prescr* settings (*prescrip* except the linear table motions) in therapy mode (8.9.8, 194). Most settings are machine *motions*, and the actual values of most settings are measured by *sensors*. The *calibration constants* are registers that are initially loaded with constants stored in the calibration database. At this writing

```

leaves == {leaf0, leaf39}
preset == leaves  $\cup$  {wedge, w-rot, filter}
motion == preset  $\cup$  {gantry, collim, turnt, lat, longit, height}
prescrip == motion  $\cup$  counter
prescr == prescrip  $\setminus$  {lat, longit, height}
sensor == setting  $\setminus$  {nfrac, dose-tot}
cal-const == {d_rate, t_fac, calvolt1, calvolt2}

```

²In the C implementation *setting* and *dose-reg* are two different enumerations, separated so we can efficiently store and index zero-based C arrays. We may add other register enumerations in the future, for example for the LCC calibration factors.

3.2 Values

The value of every item can be represented by a number.

$$VALUE == \mathbb{Z}$$

Each item will be implemented by an appropriate (possibly floating point) numeric type. In this report it is sufficient to say they are all numbers, to indicate that we can do arithmetic with them.

Each item can assume a particular range of *valid* (physically achievable) values. For example, the gantry angle can vary from 0 to 359³; the available wedge selections are *no_wedge*, 30, 45 and 60. We use *valid* to do range checking on numbers that the operator types in, and also on sensor readings, to check for faults⁴. Every setting *s* has some valid values, and there is always a minimum and a maximum valid value. We define an uninitialized or *blank* value which is not valid for any setting. For each *scale* item, there is a *tolerance* within which variations in value are acceptable.

$$\begin{array}{l} \boxed{\begin{array}{l} \textit{blank} : VALUE \\ \textit{tol} : scale \rightarrow VALUE \\ \textit{valid} : ITEM \rightarrow \mathbb{F}_1 VALUE \end{array}} \\ \hline \forall s : ITEM \bullet \textit{blank} \notin \textit{valid} s \end{array}$$

4 Prescription database

The prescription database stores patients and fields. We define a basic type for the names that identify them.

$$[NAME]$$

$$PATIENT == NAME; FIELD == NAME$$

An item's name usually corresponds to the text string that identifies it in screen displays and log files⁵.

We distinguish a special value to indicate that no name has been selected.

³In the implementation gantry angle varies from 0.0 to 359.9. Decimal fractions are not built into Z.

⁴The C implementation includes one *valid* array indexed by *setting* and another (with a different name) indexed by *dose_reg*.

⁵We define one type for both kinds of names so the same specifications (and code) can be used to handle lists of patients and fields. In the implementation, elements of *NAME* are integer indices into arrays, usually of C structures that include the name string as one member.

| $no_name : NAME$

$no_patient == no_name; no_field == no_name$

In experiment mode, we store fields under *studies* which are analogous to patients. In our model they have the same type.

$studies, patients : \mathbb{P} PATIENT$

$no_patient \notin studies \wedge no_patient \notin patients$

For each patient or study, several prescribed fields are stored⁶. We must check against delivering too many fractions or monitor units from the same field (8.9.4, 187 – 188), so the accumulated values of the counters are also stored (for patient fields only).

$ACCUMULATION == counter \rightarrow VALUE$

$PREScription == prescrip \rightarrow VALUE$

$Preset : studies \rightarrow (FIELD \leftrightarrow PREScription)$

$Prescribed : patients \rightarrow (FIELD \leftrightarrow PREScription)$

$Accumulated : patients \rightarrow (FIELD \leftrightarrow ACCUMULATION)$

$\forall s : studies \bullet no_field \notin \text{dom}(Preset s)$

$\forall p : patients \bullet no_field \notin \text{dom}(Prescribed p) \wedge \text{dom}(Prescribed p) = \text{dom}(Accumulated p)$

The *exceeded* predicate tests whether the prescribed fractional dose, total dose or number of fractions have already been delivered.

$exceeded_ : ACCUMULATION \leftrightarrow PREScription$

$\forall counters : ACCUMULATION; fields : PREScription \bullet$

$exceeded(counters, fields) \Leftrightarrow (\exists c : counter \bullet counters c \geq fields c)$

In the following discussion consider field f of patient p ; let $prescribed = Prescribed p f$ and $accumulated = Accumulated p f$. The prescription includes the number of fractions *prescribed n* and the total dose *prescribed dose_tot*. We also keep track of the number of fractions accumulated to date *accumulated n*, the number of monitor units delivered since the beginning of the day *accumulated dose* and the total number of monitor units accumulated to date *accumulated dose_tot*. Table 1 shows the settings and values pictured on each line of the field selection display (8.9.4, Fig. 83, 186).

⁶This differs from [1], which describes a single collection of experiment fields. Moreover, for each experiment field we now store the same *prescrip* settings as for therapy fields, although we only check the *preset* settings for agreement with the stored prescription

Field	<i>field</i>
Fractions	<i>prescribed n</i>
To date	<i>accumulated n</i>
MU	<i>prescribed dose</i>
Total	<i>prescribed dose_tot</i>
Expected	<i>accumulated n * prescribed dose</i>
To date	<i>accumulated dose_tot</i>

Table 1: Settings and values in the field list display

4.1 Operators

Our *OPERATOR* type includes the operator’s username and password. A special value indicates no operator has logged in. Physicists are operators who are authorized to use the equipment in its experiment mode (8.2, 170).

[*OPERATOR*]

<i>no_operator</i> : <i>OPERATOR</i>
<i>operators,physicists</i> : \mathbb{P} <i>OPERATOR</i>
<i>physicists</i> \subseteq <i>operators</i>

5 Session

In this section we model those aspects of the treatment session that are related to the prescription database. In section 9, we will combine the operations defined here with user interface operations described in section 8.

5.1 Session state

The *Session* state is determined by the treatment *mode*, the *operator* on duty, the currently selected *patient* and *field*, the accessible *names* (patients or studies), and the accessible prescribed *fields* and their *counters*. We first define *SessionVars* which declares all the state variables and provides predicates to ensure that the operator is authorized for the mode, and the names are consistent with the mode.

MODE ::= therapy | experiment

Session Vars _____

mode : MODE
operator : OPERATOR
patient : PATIENT
field : FIELD
names : P PATIENT
fields : FIELD → PRESCRIPTION
counters : FIELD → ACCUMULATION

operator = no_operator ∨ operator ∈ operators
mode = experiment ⇒ operator ∈ physicists
names = if mode = therapy then patients else studies

Next, we define two cases. When no patient is selected, no prescribed fields are accessible; no field can be selected.

NoPatient _____

Session Vars _____

patient = no_patient
field = no_field
fields = Ø
counters = Ø

When a patient is selected, that patient's fields are accessible. If a field is selected, it must be one of these.

PrescribedPatient _____

Session Vars _____

patient ≠ no_patient
patient ∈ names
field = no_field ∨ field ∈ dom fields
fields = if mode = therapy then Prescribed patient else Preset patient
mode = therapy ⇒ counters = Accumulated patient

Together these define the *Session* state.

$$\text{Session} \doteq \text{PrescribedPatient} \vee \text{NoPatient}$$

The *Session* subsystem starts up in therapy mode with no operator and no patient.

<i>InitSession</i>	_____
<i>NoPatient</i>	_____
<i>mode = therapy</i>	
<i>operator = no-operator</i>	

None of the *Session* state variables are sensor inputs; all are under program control.

5.2 Operations on *Session*

In the following subsections we model the operations on *Session*. We will put together the operations defined in different states in section 9, below.

5.2.1 Experiment mode

Physicists can toggle the session from therapy mode to experiment mode and back⁷. The user interface ensures that only physicists can invoke this operation, so there is no need here to define a total operation that describes what happens when an operator who is not a physicist attempts this operation. After switching modes, no patient (study) and no field are selected (8.9.6, 190 – 191).

<i>ExptModeS</i>	_____
<i>ΔSession</i>	_____
<i>operator ∈ physicists</i>	
<i>NoPatient'</i>	
<i>(mode', names') = if mode = therapy then (experiment, studies)</i>	
<i>else (therapy, patients)</i>	
<i>operator' = operator</i>	

5.2.2 Store Field

Store Field (8.9.5, 189 – 188) accepts a new field name, which becomes the selected field and is also added to the list of fields⁸.

⁷This is a change from the original requirements in [1], where **Experiment Mode** switches to experiment mode but **Select Patient** switches back to therapy mode.

⁸The prose [2] also requires that the new field be added to the prescription database for the current patient. We do not model this formally (in fact we model the prescription database as a constant). The precondition *patient ≠ no_patient* is not explicit in the prose.

StoreFieldS _____

$\Delta Session$

$field? : FIELD$

$prescribed' : PRESCRIPTION$

$accumulated' : ACCUMULATION$

$patient \neq no_patient$

$field' = field?$

$fields' = fields \cup \{field' \mapsto prescribed'\}$

$mode = therapy \Rightarrow counters' = counters \cup \{field' \mapsto accumulated'\}$

$mode' = mode$

$operator' = operator$

$patient' = patient$

$names' = names$

Here $prescribed'$ and $accumulated'$ are just place holders; their values are defined in the corresponding *Field* operation *StoreFieldF*.

5.2.3 Login

Login (2.5.2, 17 – 20; 8.9.1, 183) accepts a new $operator?$. The user interface ensures that the new operator is authorized.

NewOperator _____

$\Delta Session$

$operator? : OPERATOR$

$operator' = operator?$

$operator' \in operators$

There are two variations. Usually the new operator is sufficiently privileged to keep the same mode. Otherwise the session reverts to therapy mode with no patient and no field (8.9.6, 190).

Privileged

NewOperator

mode = *therapy* \vee *operator'* \in *physicists*
mode' = *mode*
patient' = *patient*
names' = *names*
field' = *field*
fields' = *fields*
counters' = *counters*

Unprivileged

NewOperator

mode = *experiment*
operator \notin *physicists*
mode' = *therapy*
NoPatient'

LoginS $\hat{=}$ *Privileged* \vee *Unprivileged*

5.2.4 Select Patient

In **Select Patient** (8.9.3, 184 – 185) the patient’s prescribed fields are loaded, but no field is selected⁹. The user interface ensures that the new patient is in the prescription database.

SelectPatientS

Δ *Session*

patient? : PATIENT

patient? \in *names*
patient' = *patient?*
field' = *no_field*
fields' = **if** *mode* = *therapy* **then** *Prescribed patient'* **else** *Preset patient'*
mode = *therapy* \Rightarrow *counters'* = *Accumulated patient'*
mode' = *mode*
operator' = *operator*
names' = *names*

⁹The prose [1] says that if the patient list is selected in experiment mode, the session reverts to therapy mode (8.9.3, 184 last paragraph). We have dropped this requirement.

5.2.5 Select Field

Select Field (8.9.4, 186 – 189) changes the current field. The user interface ensures this operation cannot occur if there is no patient, and ensures that the new field is prescribed.

<i>SelectFieldS</i>	_____
$\Delta Session$	
$field? : FIELD$	
$patient \neq no_patient$	
$field? \in \text{dom } fields$	
$field' = field?$	
$operator' = operator$	
$mode' = mode$	
$patient' = patient$	
$fields' = fields$	
$counters' = counters$	

6 Field

In this section, we look inside the machine state and deal with particular machine settings. We model operations that involve the many settings that characterize a single field.

6.1 Field state

The *Field* schema includes the state variables that represent settings for the currently selected *field* and *mode*. *Sensors* report *measured* setting values. *Prescribed* setting values are read from the prescription database.

Computed and *calibrated* item values are entered by the operator or calculated from prescribed settings and calibration constants; these are stored in *registers*. Certain *calibration constants* are stored in files (8.9.13, 213 first full paragraph; 215 last paragraph). *Counters* hold setting values that are *accumulated* over successive runs. For example, the *dose* prescribed for a single fraction may be have to be delivered in two or more treatment runs.

Some settings that do not match their prescribed values can be *overridden* by the operator (8.4, 175 second paragraph; 8.8.1, 181). It is necessary to store the value of each setting

when it is overridden (see the requirement in the last paragraph under “override” on p. 181). Only settings that are prescribed can be overridden.

| *cal_factor* : *cal_const* → *VALUE*

Field _____

<i>prescribed</i> : <i>PREScription</i>
<i>accumulated</i> : <i>ACCUMULATION</i>
<i>measured</i> : <i>sensor</i> → <i>VALUE</i>
<i>overridden</i> : <i>prescr</i> →→ <i>VALUE</i>
<i>computed, calibrated</i> : <i>dose_reg</i> → <i>VALUE</i>

The *measured* settings are read from sensors so here we cannot write any predicates that constrain them.

6.2 Relation to *Session* state

A few operations on *Field* read the *mode* and *field* state variables declared in *Session*. In therapy mode, the *prescribed* settings in the *Field* state are those from the prescription database entry for the currently selected *mode* and *field* in the *Session* state. (In experiment mode the prescribed settings are also loaded from the prescription database but may be changed subsequently. In therapy mode the counters are loaded from the prescription database when the field is selected but may be changed subsequently. See section 6.4.2).

PrescribedField _____

Field

Session

<i>field</i> ≠ <i>no_field</i>
<i>mode</i> = <i>therapy</i> ⇒ <i>prescribed</i> = <i>fields field</i>

When no field has been selected, prescribed settings and counters have no values and the computed settings dose and time indicate no dose. (8.9.7, 192, second paragraph after the bullets). No settings are overridden (8.9.8, 194; 8.9.9, 196; 8.9.10, 198).

$$\begin{aligned} \textit{no_prescrip} &== (\lambda p : \textit{prescrip} \bullet \textit{blank}) \\ \textit{no_counter} &== (\lambda c : \textit{counter} \bullet \textit{blank}) \\ \textit{no_dose_reg} &== (\lambda d : \textit{dose_reg} \bullet \textit{blank}) \\ \textit{no_dose} &== \{p_dose \mapsto \textit{blank}, p_time \mapsto \textit{blank}\} \end{aligned}$$

<i>NoFieldF</i>	_____
<i>Field</i>	_____
<i>prescribed</i>	= <i>no_prescrip</i>
<i>accumulated</i>	= <i>no_counter</i>
<i>no_dose</i>	⊆ <i>computed</i>
<i>overridden</i>	= \emptyset

$$NoFieldS \triangleq [Session \mid field = no_field]$$

$$NoField \triangleq NoFieldF \wedge NoFieldS$$

FieldSession expresses the combined invariant:

$$FieldSession \triangleq PrescribedField \vee NoField$$

6.3 Initialization

Field begins with no field. The calibration factors are initialized with the constants on file (8.9.13, 213, second paragraph after bullets) and the other registers hold no values.

<i>InitField</i>	_____
<i>NoFieldF</i>	_____
<i>computed</i>	= <i>calibrated</i> = <i>no_dose_reg</i> ⊕ <i>cal_factor</i>

6.4 Operations on *Field*

In the following subsections we model the operations on *Field*. We will put together the operations defined in different states in section 9, below.

6.4.1 Select Patient

SelectPatient also affects *Field*: when a patient is first selected, there is no field.

SelectPatientF _____

$\Delta Field$

NoFieldF'

$computed' = computed \oplus no_dose$

$calibrated' = calibrated$

6.4.2 Select Field

When a new field is selected, its prescribed settings are loaded and no settings are overridden. This operation requires read-only access to the *fields* state variable in the *Session* schema.

NewFieldF _____

$\Delta FieldSession$

$prescribed' = fields\ field'$

$overridden' = \emptyset$

There are two variants of *SelectField*. Experiment mode is much simpler because there is no prescribed dose. The prescribed settings are loaded. The dose and time do not change (8.9.11, 202, second paragraph from bottom).

SelectExptFieldF _____

NewFieldF

$mode = experiment$

$computed' = computed$

$calibrated' = calibrated$

Selecting rectangular fields in experimental mode (8.9.4, 188 – 189) is not modelled formally.

In therapy mode, the dose for the treatment run and the treatment backup time are calculated. Treatment backup time is calculated from the dose and two calibration factors, the machine's nominal dose rate *computed d_rate* and the treatment time factor *computed t_fac* (8.9.11, 200, last paragraph; 202, second paragraph; 8.9.13, 213, first two paragraphs after bullets)¹⁰.

¹⁰The backup time is given by $t_{backup} = factor * dose / rate$. For example with prescribed dose 100.0 MU, dose rate 50.0 MU/min and factor 1.50 the backup time is 3.00 minutes. We do not attempt to model this floating-point calculation in Z.

SETTING	PRESKR	PRESET	ACCUM
DOSE A	<i>prescribed dose</i>	<i>computed p_dose</i>	<i>measured dose</i>
DOSE B	<i>prescribed dose</i>	<i>computed p_dose</i>	<i>measured doseB</i>
TIME	<i>calibrated p_time</i>	<i>computed p_time</i>	<i>calibrated e_time</i>

Table 2: Settings and values in the dosimetry display

DOSE == VALUE; RATE == VALUE; FACTOR == VALUE; TIME == VALUE

$$\begin{array}{|l}
 \hline
 t_{\text{backup}} : (DOSE \times RATE \times FACTOR) \rightarrow TIME \\
 \hline
 \forall d : \text{valid dose}; r : \text{valid } d_{\text{rate}}; f : \text{valid } t_{\text{fac}} \bullet \\
 (d, r, f) \in \text{dom } t_{\text{backup}} \wedge t_{\text{backup}}(d, r, f) \in \text{valid } p_{\text{time}} \\
 \hline
 \end{array}$$

We keep track of the number of monitor units delivered since the beginning of the day *accumulated dose*. When the prescribed field settings are loaded, the computed dose is adjusted to deliver the remaining daily dose. This makes it easy to set up another treatment run for the same field if the earlier attempts had to be interrupted for any reason, or were used to make a port film. The treatment backup time is calculated from this adjusted dose, not the prescribed dose.

The adjusted dose and corresponding backup time are stored in *computed p_dose* and *calibrated p_time* (*computed p_dose* may differ from *prescribed dose*). There is also a register *computed p_time* where the user may optionally enter a backup time different than *calibrated p_time* (section 6.4.3, below). Table 2 shows the settings and values pictured on the dosimetry display (8.9.11, Fig. 8.8, 199; Fig. 8.9, 203; Fig 8.10, 207; Fig. 8.11, 208).

$$\begin{array}{|l}
 \hline
 DoseTime \\
 \hline
 \Delta Field \\
 \hline
 (\text{let } t == t_{\text{backup}}(\text{computed}' p_{\text{dose}}, \text{computed}' d_{\text{rate}}, \text{computed}' t_{\text{fac}}) \bullet \\
 \quad \text{calibrated}' = \text{calibrated} \oplus \{ p_{\text{time}} \mapsto t \}) \\
 \quad \text{computed}' p_{\text{time}} = \text{calibrated}' p_{\text{time}} \\
 \quad \{ p_{\text{dose}}, p_{\text{time}} \} \triangleleft \text{computed}' = \{ p_{\text{dose}}, p_{\text{time}} \} \triangleleft \text{computed} \\
 \hline
 \end{array}$$

$$\begin{array}{|l}
 \hline
 NewTherapyField \\
 \hline
 NewFieldF \\
 DoseTime \\
 \hline
 mode = therapy \\
 accumulated' = counters field' \\
 \hline
 \end{array}$$

There are two cases. The normal case occurs when the user interface confirms that the prescribed fractional dose, total dose and number of fractions are not yet *exceeded*. The dose is read from the prescription, and no settings are overridden (8.9.4, 187).

<i>SelectTherapyFieldF</i>	_____
<i>NewTherapyField</i>	
	<i>computed'</i> $p_dose = \text{prescribed dose} - \text{accumulated dose}$
	<i>overridden'</i> = \emptyset

Together these make the simple case

$$\text{SelectSimpleFieldF} \doteq \text{SelectExptFieldF} \vee \text{SelectTherapyFieldF}$$

The other case occurs when the user interface acquires the preset dose from the operator (often when one or more of the counter settings is *exceeded*). If this differs from the prescribed dose then dose is overridden, and any exceeded settings are also overridden (8.9.4, 188).

<i>SelectComplexFieldF</i>	_____
<i>NewTherapyField</i>	
<i>dose?</i> : VALUE	
	<i>computed'</i> $p_dose = \text{dose?}$
	(let $ovr == (\lambda c : \text{counter} \mid \text{accumulated}' c \geq \text{prescribed}' c \bullet \text{accumulated } c)$ •
	<i>overridden'</i> = if $\text{dose?} = \text{prescribed}' \text{ dose}$
	then ovr else $ovr \cup \{\text{dose} \mapsto \text{dose?}\}$

Here we have made a few small changes from the prose requirements. According to the prose (8.9.4, 187 – 188), the **Select Field** operation includes a dialog with the operator to enter a new dose or treatment time in some cases. In our formal specification it is necessary for the operator to explicitly select the **Edit** operation after **Select Field** in order to enter a new dose or treatment time. These minor adjustments achieve the intent of the prose and simplify the program. As required by the prose, our *SelectComplexFieldF* overrides exceeded settings (after operator confirmation, enforced by the user interface)¹¹.

6.4.3 Edit setting

The edit operation updates a *prescribed* or *computed* item value.

¹¹We also considered the slightly simpler alternative of omitting the operator confirmation and leaving $\text{overridden} = \emptyset$ in the *exceeded* case. In that alternative, the *Intlk* subsystem (section 7) would make the offending settings *not-ready* to prevent the field being delivered unless the operator explicitly edits or overrides those settings.

EditF

$\Delta Field$

$item? : ITEM$

$value? : VALUE$

$accumulated' = accumulated$

$calibrated' = calibrated$

The prose actually describes four **Edit** operations. Some features are common to all. The first variation is for preset settings; the user interface ensures this can be invoked in experiment mode only (8.8.1, 180). The prescribed value is changed, and that setting is no longer overridden.

EditPresetF

EditF

$item? \in preset$

$prescribed' = prescribed \oplus \{item? \mapsto value?\}$

$overridden' = \{item?\} \triangleleft overridden$

$computed' = computed$

The second variation is for calibration factors; again, the user interface only provides this in experiment mode (8.9.3, 215). Calibration factors that users can edit are modelled as *computed* settings in *registers*. Calibration factors are never considered overridden.

EditCalF

EditF

$item? \in dose_reg \setminus \{p_dose, p_time\}$

$computed' = computed \oplus \{item? \mapsto value?\}$

$prescribed' = prescribed$

$overridden' = overridden$

The third variation is for dose (8.9.11, 201–202). The computed (not prescribed) value is changed, and the dose is considered overridden (8.9.4, 188; 8.9.11, 202). The treatment times are recalculated.

EditDoseF _____

EditF

DoseTime

item? = p_dose

computed' p_dose = value?

overridden' = overridden ⊕ {dose ↦ value?}

prescribed' = prescribed

The fourth and last variation is treatment backup time, which can be edited in both modes (8.9.11, 202). Here again the computed value is changed; time is not a prescribed setting, so it cannot be overridden.

EditTimeF _____

EditF

item? = p_time

computed' = computed ⊕ {p_time ↦ value?}

prescribed' = prescribed

overridden' = overridden

Here is the combined operation:

EditSettingF \cong *EditCalF* \vee *EditPresetF* \vee *EditDoseF* \vee *EditTimeF*

EditSettingF is not a total operation (it does not handle all possible values of *ITEM*) but the user interface ensures that its preconditions are always satisfied.

We now provide the *EditDoseF* and *EditTimeF* operations instead of the dialog after **Select Field** proposed in [2] (8.9.4, 187 – 188).

6.4.4 Override

Certain items can be overridden.

OverF _____

Δ *Field*

item? : ITEM

prescribed' = prescribed

accumulated' = accumulated

computed' = computed

calibrated' = calibrated

We add a newly overridden setting and its currently measured value to the *overridden* function (8.4, 175 second paragraph; 8.8.1, 181). If the setting is already overridden, the override is cancelled.

$\begin{aligned} &\text{OverrideSetting} \\ &\text{OverF} \end{aligned}$
$\begin{aligned} &\text{item?} \in \text{prescr} \\ &\text{overridden}' = \\ &\quad \text{if item?} \notin \text{dom overridden} \\ &\quad \quad \text{then overridden} \oplus \{\text{item?} \mapsto \text{measured item?}\} \\ &\quad \quad \text{else } \{\text{item?}\} \triangleleft \text{overridden} \end{aligned}$

Dose and time are special cases; overriding either makes dose overridden with its accumulated (not measured) value as the overridden value. The counters total dose *dose_tot* and number of fractions *nfrac* can only be overridden (after operator confirmation) as part of the *SelectFieldF* operation.

$\begin{aligned} &\text{OverrideDose} \\ &\text{OverF} \end{aligned}$
$\begin{aligned} &\text{item?} \in \{p_dose, p_time\} \\ &\text{overridden}' = \\ &\quad \text{if dose} \notin \text{dom overridden} \\ &\quad \quad \text{then overridden} \oplus \{\text{dose} \mapsto \text{accumulated dose}\} \\ &\quad \quad \text{else } \{\text{dose}\} \triangleleft \text{overridden} \end{aligned}$

$$\text{OverrideF} \hat{=} \text{OverrideSetting} \vee \text{OverrideDose}$$

6.4.5 Store Field

This operation (8.9.5, 189 – 190) makes the prescribed settings equal to the actual machine settings, except there is no prescribed dose and the number of fractions is set to one. The accumulators are reset to zero.

$$\text{zero_counter} == (\lambda c : \text{counter} \bullet 0)$$

<i>StoreFieldF</i>	_____
$\Delta FieldSession$	_____
<i>computed'</i> = <i>computed</i> \oplus <i>no_dose</i>	
<i>prescribed'</i> = <i>prescribed</i> \oplus (<i>prescrip</i> \triangleleft <i>measured</i>) \oplus <i>no_counter</i> \oplus { <i>nfrac</i> \mapsto 1}	
<i>accumulated'</i> = <i>zero_counter</i>	
<i>overridden'</i> = \emptyset	
<i>calibrated'</i> = <i>calibrated</i>	

6.4.6 Experiment Mode

This operation toggles modes with no field. There are no dose and time (8.9.11, 202, second paragraph from bottom).

<i>ExptModeF</i>	_____
$\Delta Field$	_____
<i>NoFieldF'</i>	
<i>computed'</i> = <i>computed</i> \oplus <i>no_dose</i>	
<i>calibrated'</i> = <i>calibrated</i>	

6.5 Calibration factors

6.5.1 Dosimetry calibration

Dosimetry calibration factors, including the dose rate and treatment time factor used to calculate the backup time, appear on the **Dosimetry Calibration** display (8.9.13, 213 – 214)¹². Table 3 shows part of a possible design for this display. The *calibrated* values in the left column are read from files or measured by sensors, while the *computed* values in the right column are computed by the control program or entered by the operator using the *EditCalF* operation.

The pressure-temperature correction factors are used to adjust the standard calibration voltages for the dosimetry system (8.9.13, 213 – 215). The *calibrated calvolt1* and *calibrated calvolt2* represent the standard calibration voltages on file (8.9.13, 213, second paragraph from bottom), while *computed calvolt1* (etc.) represent the calibration voltages actually in effect,

¹²Called **Cal Factors** in [2], since renamed to distinguish it from the forthcoming **LCC Calibration** etc.

	MEASURED/CALIBRATED	ADJUSTED
P/T MODE	<i>computed pt_mode (automatic/manual)</i>	
PRESSURE	<i>calibrated press</i>	<i>computed press</i>
TEMPERATURE	<i>calibrated temp</i>	<i>computed temp</i>
P/T CORR.	<i>calibrated pt_factor</i>	<i>computed pt_factor</i>
CALVOLT 1	<i>calibrated calvolt1</i>	<i>computed calvolt1</i>
CALVOLT 2	<i>calibrated calvolt2</i>	<i>computed calvolt2</i>
DOSE RATE	<i>calibrated d_rate</i>	<i>computed d_rate</i>
TIME FACTOR	<i>calibrated t_fac</i>	<i>computed t_fac</i>

Table 3: Dosimetry calibration display

which are obtained by adjusting the standard calibration voltage by a barometric pressure/temperature correction factor (8.9.13, 213 bottom paragraph, 214 top paragraph)¹³.

PRESSURE == VALUE; TEMPERATURE == VALUE

$$\begin{aligned} \text{pt_formula} : (\text{PRESSURE} \times \text{TEMPERATURE}) &\rightarrow \text{FACTOR} \\ \forall p : \text{valid press}; t : \text{valid temp } \bullet \\ (p, t) \in \text{dom pt_formula} \wedge \text{pt_formula}(p, t) &\in \text{valid pt_factor} \end{aligned}$$

The *computed press* and *computed temp* are the pressure and temperature entered by the operator, and while *calibrated temp* and *calibrated press* are measured continuously by sensors. The *computed pt_factor* stores the *barometric pressure/temperature correction factor of the day* calculated from the readings entered by the operator (8.9.13, 214, third paragraph), while *calibrated pt_factor* stores the *automatic pressure/temperature correction factor* calculated from sensor readings (8.9.13, 214, fourth paragraph). The *pressure-temperature interlock* (section 7) accounts for the possibility that the pressure or temperature values might be invalid or expired.

The operator sets *computed pt_mode = automatic* to use the *automatic pressure/temperature correction factor*, and *computed pt_mode = manual* to use the correction factor that is based on the manually entered values (8.9.13, 214, fifth paragraph).

automatic, manual : *VALUE*

The *ScanPT* operation computes the correction factors and updates the registers with the new values.

¹³The pressure-temperature factor is given by $pt_factor = (press/1013) \times (295/(temp + 273))$, where *press* and *temp* are in mbar and deg. C, respectively. We do not attempt to model this floating-point calculation in Z.

ScanPT _____
 $\Delta Field$

```
calibrated' pt_factor = pt_formula(calibrated press, calibrated temp)
computed' pt_factor = pt_formula(computed press, computed temp)

(let pt_corr == if computed pt_mode = automatic
    then calibrated' pt_factor else computed' pt_factor •
    computed' calvolt1 = pt_corr * calibrated calvolt1 ∧
    computed' calvolt2 = pt_corr * calibrated calvolt2)

{pt_factor} ⊑ calibrated' = {pt_factor} ⊑ calibrated
{pt_factor, calvolt1, calvolt2} ⊑ computed' = {pt_factor, calvolt1, calvolt2} ⊑ computed

prescribed' = prescribed
accumulated' = accumulated
overridden' = overridden
```

ScanPT is scheduled by the control program itself; it is not invoked by the user.

7 Software interlocks and status flags

(To come)

8 User interface

The user may provide input at the workstation at any time (by typing, pressing function keys or cursor arrow keys — in our implementation we do not use the mouse). We model each keystroke and the actions it invokes as an *Event* that accepts an *input?* that may change the *Console* state.

<i>Event</i>	_____
$\Delta \text{Console}$	
<i>input? : INPUT</i>	

We do not attempt to formalize any “look and feel” aspects of the user interface, such as the appearance of the display. They are already described in sufficient detail in [2], chapters 2 and 8.

INPUT is the set of inputs (keypresses) the user can provide¹⁴. Here is the list of inputs at this writing.

```
INPUT ::= filter_wedge | leaf_collim | dose_intlk | gantry_psa | dose_cal |  
        startup | help | messages | select_patient | select_field | field_summary |  
        login | edit_setting | edit_dose_reg | log_message | store_field | override_cmd |  
        cancel_run | password | auto_setup | expt_mode | cancel | refresh | shutdown |  
        select | ret | character | backspace | delete_key |  
        left_arrow | right_arrow | up_arrow | down_arrow | ignored
```

Many operations are invoked by pressing keys, so it is often convenient to identify operations with the corresponding input. Therefore we assign them to the same type. Here is the list of operations at this writing.

<i>OP : P INPUT</i>	_____
$OP = \{filter_wedge, leaf_collim, dose_intlk, gantry_psa, dose_cal,$ $startup, help, messages, select_patient, select_field, field_summary,$ $login, edit_setting, edit_dose_reg, log_message, store_field, override_cmd,$ $cancel_run, password, auto_setup, expt_mode, cancel, refresh, shutdown, select\}$	

The user interface shows many displays, for example the login display (Fig 8.1, 178), the patient list display (Fig. 8.2, 185), the leaf collimator display (Fig. 8.7, 197) etc. The operator can choose any display by pressing a key, so we can identify displays with these operations.

¹⁴In the implementation, inputs are X window system events and the values of *INPUT* correspond to X keysyms [5].

$DISPLAY : \mathbb{P} OP$
$DISPLAY = \{filter_wedge, leaf_collim, dose_intlk, gantry_psa, dose_cal, startup, help, messages, select_patient, select_field, field_summary, login\}$

8.1 Console state

This section describes the variables in the *Console* state.

The first variable indicates the mode of *interaction*. If no interaction is in progress the console is *available*, or there may be a *dialog* in progress where the user is typing text into a dialog box, or there may be a *menu* displayed, or the user may be asked to *confirm* some operation by providing a yes/no answer (this mode can also be used to present informational messages).

$$INTERACTION ::= available \mid dialog \mid menu \mid confirm$$

The *op* variable keeps track of which top-level operation (described in [2]) is underway.

The *display* variable indicates which of the screen designs pictured in the informal specification is currently visible on the display. The *display* variable determines which items appear and helps determine which operations are available.

The *item* state variable holds the item which the operator has selected from a tabular display, for example the setting which the operator is editing.

The *nlist* state variable holds the list of names (of patients or fields) that appear on a list display, and *list-item* indicates the currently selected name.

The *menu-item* state variable holds the index of the current menu selection (a small integer).

$$| nmax : \mathbb{N}$$

$$SELECTION == \{i : \mathbb{N} \mid i \leq nmax\}$$

The *buffer* state variable models the (possibly incomplete) string that the user edits in dialog mode.

$$[STRING]$$

| *empty* : STRING

The *keyswitch* must be unlocked to allow the console to be used (8.7, 179).

KEYSWITCH ::= *locked* | *unlocked*

Some operations are available only when a treatment is being set up, and are locked out while a treatment *run* is in progress (8.8.2, 183).

RUN ::= *setup* | *running*

The *keyswitch* and *run* variables depend on sensor inputs; they are not constrained here.

Together, these variables describe the state of the user interaction.

Console _____

keyswitch : *KEYSWITCH*
run : *RUN*
display : *DISPLAY*
op : *OP*
interaction : *INTERACTION*
item : *ITEM*
nlist : \mathbb{P} *NAME*
list_item : *NAME*
menu_item : *SELECTION*
buffer : STRING

When the control program starts up, the login process begins (section 8.3.13)¹⁵.

InitConsole _____

Console _____

op = *login*
display = *login*
interaction = *dialog*
buffer = *empty*

8.2 Elements of user interaction

All user interactions are built up from a few elements. In this section we define the constants, states and operations that serve as building blocks.

¹⁵When the implementation starts up, the *startup* screen appears first. The login process does not begin until the various *Init...* conditions are established. We do not model this formally.

The *caption* type models messages or other output to the operator that appear temporarily at the console (in dialog boxes or perhaps even from the speaker, see 2.2.3, 9). Captions are distinguished from *log messages* which appear in a different location on the console and are also stored in log files along with timestamps other information (2.2.4, 9).

$[CAPTION, MESSAGE]$

Ignore is the default do-nothing operation that is invoked when a key is pressed but the preconditions for the associated operation are not satisfied. *Ignore* does not change the state, but issues an alert (such as sounding the workstation bell) to notify the user that the input was received but the operation is not enabled.

| *alert* : *CAPTION*

<i>Ignore</i>	_____
<i>Event</i>	
$\exists Console$	
<i>caption!</i> : <i>CAPTION</i>	
<i>caption!</i> = <i>alert</i>	_____

The keyswitch must be unlocked for any operation to occur. When the keyswitch is locked, input is ignored:

$Unlocked \doteq [Console \mid keyswitch = unlocked]$

$EventUnlocked \doteq Event \wedge Unlocked$

Many operations are invoked by pressing the *select* key.

$Select \doteq [EventUnlocked \mid input? = select]$

It is convenient to describe the operations that can occur in each of the interaction modes. Each mode is described in a following subsection.

8.2.1 Available

Most of the top-level operations described in [2] can only be selected when the console is available.

$Available \doteq [Console \mid interaction = available]$

<i>Op</i>
<i>EventUnlocked</i>
<i>Available</i> <i>input?</i> $\in OP$

Certain operations have stronger preconditions: they cannot occur when a run is in progress (8.8.2, 183). A few operations occur only when a run is in progress (8.9.11, 209-210).

$$Setup \hat{=} [Available \mid run = setup]$$

$$Running \hat{=} [Available \mid run = running]$$

When the console is available, the user may select a new *display*. The console remains available.

<i>SelectDisplay</i>
<i>Op</i>
<i>input?</i> $\in DISPLAY$
<i>display' = input?</i>
<i>op' = display'</i>
<i>Available'</i>

SelectDisplay operations may change *item* and *list-item* (see below) but do not change other state variables (for brevity we omit the $x' = x$ “nothing changes” predicates).

When an interaction is in progress, the console is *Engaged*. The *Done* operation schema describes what happens when an interaction completes: the console returns to the *Available* state, and *op* returns to its value when the display was selected.

$$Engaged \hat{=} [Console \mid interaction \neq available]$$

<i>Done</i>
<i>EventUnlocked</i>
<i>Engaged</i>
<i>op' = display</i>
<i>display' = display</i>
<i>Available'</i>

The *Cancel* operation is used to end an interaction without making permanent changes to the underlying machine state.

$$Cancel \hat{=} [Done \mid input? = cancel]$$

8.2.2 Lists

Certain displays show a list of names (patients or fields). When a list display is selected, $nlist$ is loaded, and the default $list_item$ is assigned. If the list is not empty, the $List$ state results (the patient list might be empty if there are no patients on file; the field list is always empty when there is no patient, and may be empty if there are no fields on file for the selected patient).

$$\boxed{\begin{array}{l} list : \mathbb{P} DISPLAY \\ default_name : \mathbb{P}_1 NAME \rightarrow NAME \\ \hline \forall list : \mathbb{P}_1 NAME \bullet default_name list \in list \end{array}}$$

$$List \triangleq [Available \mid display \in list \wedge nlist \neq \emptyset \wedge list_item \in nlist]$$

$$\boxed{\begin{array}{l} SelectList \\ \hline SelectDisplay \\ \hline input? \in list \\ ((nlist = \emptyset \wedge list_item' = no_name) \\ \vee (List' \wedge list_item' = default_name nlist')) \end{array}}$$

Here $display \in list \wedge nlist \neq \emptyset$ distinguishes the $List$ state, and this test occurs explicitly in the implementation. In contrast, $list_item \in nlist$ is an invariant. It need not be coded as an explicit test but it must be maintained or else the implementation might abort (because $list_item$ is used as an index into $nlist$).

The console indicates $list_item$ (for example by placing a highlight or cursor over that name in the list). Subsequently the user can choose a new name from the list by using the up and down-arrow keys. The function $aname$ calculates the new name by “dead reckoning” from the old name, the list, and the arrow key (it is not necessary for the program to poll the console for the cursor position). The list remains visible.

$$v_arrow == \{ up_arrow, down_arrow \}$$

$$\boxed{\begin{array}{l} aname : (v_arrow \times NAME \times \mathbb{P}_1 NAME) \rightarrow NAME \\ \hline \forall a : v_arrow; n : NAME; list : \mathbb{P}_1 NAME \bullet aname(a, n, list) \in list \end{array}}$$

$$Continue \triangleq [\Delta Console \mid interaction' = interaction \wedge op' = op \wedge display' = display]$$

GetListArrow —————
EventUnlocked
 $\Delta List$

input? $\in v_arrow$
 $list_item' = aname(input?, list_item, nlist)$
Continue

This is a *Continue* operation that does not change *interaction*, *op*, or *display*. Here *list_item* is the only state variable that changes. We do not completely specify *default_name* and *aname*; we leave that to the implementation. Here we merely provide the predicates needed to ensure that the implementation does not abort.

The user presses the *select* key to choose the current *list_item* for some purpose. The selection is logged; *nmessage* converts the name to a log message.

| *selected-msg* : NAME \rightarrow MESSAGE

SelectName —————
Select
 $name! : NAME$
 $message! : MESSAGE$

List
 $name! = list_item$
 $message! = selected_msg name!$

GetListArrow and *SelectName* are not total operations; they do not handle the case where *nlist* = \emptyset . The latter case is handled by a default do-nothing operation, *IgnoreOthers* (section 8.4).

8.2.3 Tables

Certain displays show a table of items (settings for one subsystem, calibration factors etc.). The constant *table_items* tells which items on each table can be selected for editing or overriding (additional items may be displayed as well). When a tabular display is selected, the default *item* is assigned, and the *Table* state results.

| *table* : $\mathbb{P} DISPLAY$

$default_item : table \rightarrow ITEM$
$table_items : table \rightarrow \mathbb{P}_1 ITEM$
$\forall d : table \bullet default_item d \in table_items d$

$Table \hat{=} [Available \mid display \in table \wedge item \in table_items display]$

$SelectTable$
$SelectDisplay$
$input? \in table$
$item' = default_item display'$
$Table'$

Subsequently the user can indicate a new item on the table by using all four arrow keys.

$arrow == \{right_arrow, left_arrow\} \cup v_arrow$

$asetting : (arrow \times ITEM \times table) \rightarrow ITEM$
$\forall a : arrow; s : ITEM; d : table \bullet asetting(a, s, d) \in table_items d$

$GetSettingArrow$
$EventUnlocked$
$\Delta Table$
$input? \in arrow$
$item' = asetting(input?, item, display)$
$Continue$

Here $item$ is the only state variable that changes.

Items can be selected from tabular displays for editing or overriding. Editing or overriding is only enabled in the *Setup* state (when a treatment run is not in progress, see 8.8.2, 183). Pressing the *select* key when certain tabular displays are present invokes an editing operation: *edit_setting* if the selected item is a *setting* and *edit_dose_reg* if it is a *dose_reg* (notice that here op' is not the same as $input?$). Therefore it is necessary to separate *setting* and *dose_reg* items on different tables¹⁶.

¹⁶Because the implementation cannot distinguish *setting* from *dose_reg* based on *item* alone; *item* values are just C `enum` values (integers).

<i>setting-table</i> , <i>dose-reg-table</i> : \mathbb{P} table
$\forall d : setting_table \bullet table_items d \subseteq setting$
$\forall d : dose_reg_table \bullet table_items d \subseteq dose_reg$

<i>SelectItem</i> _____
<i>Select</i>
<i>Setup</i>
<i>Table</i>
<i>item' = item</i>
$(op' = edit_dose_reg \wedge display \in dose_reg_table \vee$
$op' = edit_setting \wedge display \in setting_table)$

The postcondition here implies *Editing*, the invariant of the editing state¹⁷.

<i>Editing</i> _____
<i>Console</i>
<i>interaction</i> $\in \{dialog, menu\}$
$(op = edit_dose_reg \wedge item \in dose_reg \vee$
$op = edit_setting \wedge item \in setting)$

The *Setup* precondition of *SelectItem* prevents the console entering the *Editing* state when a run is in progress. Other mechanisms prevent the machine from beginning a run while in the *Editing* state.

8.2.4 Confirm

Confirm interactions present a query (“Are you sure …?”) and wait for the user to provide a yes/no answer, indicated by the *select* or *cancel* keys (for example see 8.9.11, 210). Each *Confirm* operation presents a confirmation box (a sort of dialog box) with a *caption* that identifies the operation, and the *query*. The display under the confirmation box does not change.

Confirm $\hat{=} [Console \mid interaction = confirm]$

| *ocaption* : $OP \rightarrow CAPTION$

¹⁷The implementation uses *op* to determine whether *item* is an index into *setting* or *dose-reg*.

<i>ConfirmOp</i>	—————
<i>Op</i>	
<i>caption!</i> , <i>query!</i> : CAPTION	
<hr/>	
<i>caption!</i> = <i>o</i> <i>caption op'</i>	
<i>display'</i> = <i>display</i>	
<i>Confirm'</i>	

$$\text{Accept} \text{Confirm} \hat{=} \text{Confirm} \wedge \text{Select} \wedge \text{Done}$$

8.2.5 Menu

When the console is *Available* the user can invoke a menu, then make a selection from the menu. Each menu includes a caption and a list of menu entries. The display does not change.

<i>default_selection</i> : SELECTION	
--------------------------------------	--

$$\text{Menu} \hat{=} [\text{Editing} \mid \text{interaction} = \text{menu}]$$

<i>MenuOp</i>	—————
<i>Op</i>	
<i>caption!</i> : CAPTION	
<i>menu!</i> : iseq CAPTION	
<hr/>	
<i>menu_item'</i> = <i>default_selection</i>	
<i>display'</i> = <i>display</i>	
<i>Menu'</i>	

Here *op* also changes; the other state variables retain the same values.

Menus are used to choose new values for *selection* items; valid *selection* values are small integers. Combining *MenuOp* with *SelectItem* yields the *MenuEdit* operation. The menu shows the item name and a sequence of descriptive strings indexed by the corresponding item values. Here again, the *Editing* postcondition of *SelectItem* guarantees that *op* can be used to help look up *selection-values item*.

<i>setting_info_name</i> : ITEM \rightarrow CAPTION	
<i>setting_value</i> : selection \rightarrow iseq CAPTION	
<hr/>	
$\forall s : \text{selection} \bullet \text{dom}(\text{setting_value } s) = \text{valid } s$	

MenuEdit

MenuOp

SelectItem

item ∈ *selection*

caption! = *setting_info_name item*; *menu!* = *setting_value item*

There are functions to return the default menu selection and the new selection after each up or down-arrow keypress.

amenu : $(v_arrow \times SELECTION \times selection) \rightarrow SELECTION$

$\forall s : selection \bullet (\text{let } n == \#\text{(valid } s\text{)} \bullet$

$\forall a : v_arrow; i : SELECTION \bullet \text{default_selection} \leq n \wedge amenu(a, i, s) \leq n)$

GetMenuArrow

EventUnlocked

$\Delta Menu$

input? ∈ *v_arrow*

menu_item' = *amenu(input?, menu_item, item)*

Continue

Here *item* is the only variable that changes.

The user presses *select* to accept the current menu item and the console becomes available again.

AcceptMenu ≡ *Menu* ∧ *Select* ∧ *Done*

MenuSettingC

AcceptMenu

item! : *ITEM*

value! : *VALUE*

Editing

item! = *item*

value! = *menu_item*

8.2.6 Dialog

When the console is *Available* the user can begin a dialog, then type and edit text in a dialog box. The dialog box contains a *caption* and a *prompt* that may include the values of other state variables. The display under the dialog box does not change.

$$Dialog \hat{=} [Console \mid interaction = dialog]$$

DialogOp

Op

caption!, *prompt!* : CAPTION

caption! = *o**caption op'*

display' = *display*

Dialog'

Here only *interaction*, *buffer*, and *op* change. The *buffer* may be emptied, or may be filled with a convenient default value. We'll describe changes to *op* later, with each dialog operation.

The console remains in *Dialog* while the user types and edits. The *GetChar* operation gets a single character and updates the buffer as described by the *modify* function (append printing characters to the end of *buffer*, and do the appropriate things with editing characters).

| *CHAR* : $\mathbb{P} INPUT$

| *modify* : $(STRING \times CHAR) \rightarrow STRING$

GetChar

EventUnlocked

$\Delta Dialog$

input? $\in CHAR$

buffer' = *modify(buffer, input?)*

Continue

Here *buffer* is the only variable that changes.

When a dialog is done, the dialog box disappears and the console becomes available again. At any time the user can *cancel* the dialog and discard the input. To submit the input, the user presses a *terminator* key; the program can *Accept* the input or *Reprompt* (the user may also *Cancel* the dialog).

| *terminator* : $\mathbb{P} INPUT$

Accept

Done

Dialog

input? $\in terminator$

Reprompt _____

EventUnlocked

Δ *Dialog*

input? \in *terminator*

buffer' = *empty*

Continue

Here again, *buffer* is the only variable that changes.

Dialogs are frequently used to edit item values. Combining *DialogOp* with *SelectItem* yields the *DialogEdit* operation. Dialog box editing begins if the selected item is not a *selection* (does not have just a few discrete values). The program captions the dialog box with the item name and the minimum and maximum valid item values¹⁸.

MIN == *VALUE*; *MAX* == *VALUE*

| *setting_info* : *ITEM* \times *MIN* \times *MAX* \rightarrow *CAPTION*

DialogEdit _____

DialogOp

SelectItem

item \notin *selection*

prompt! = (**let** *v* == *valid item* • *setting_info*(*item*, *min v*, *max v*))

The implementation uses the value of *op* guaranteed by the *Editing* postcondition of *SelectItem* to look up *valid item*; there are separate *valid* arrays for *dose_reg* and *setting*.

When the user presses a terminator key, the program attempts to convert the buffer contents to a (numeric) value (non-numeric strings are always converted to an out-of-range value). If the conversion succeeds and the value is valid for the item, the dialog ends and the item and its value are reported; otherwise, the program reprompts.

| *sval* : *STRING* \rightarrow *VALUE*

EditSettingC _____

Accept

item! : *ITEM*

value! : *VALUE*

Editing

item! = *item*

(**let** *v* == *sval buffer* • *v* \in *valid item* \wedge *value!* = *v*)

¹⁸The dialog box caption also includes the units, but we do not model this formally.

$$\text{InvalidSetting} \hat{=} [\text{Reprompt} \mid \text{Editing} \wedge \text{sval buffer} \notin \text{valid item}]$$

In the implementation it is convenient to combine these two operations¹⁹.

$$\text{EditOrInvalidSetting} \hat{=} \text{EditSettingC} \vee \text{InvalidSetting}$$

8.2.7 Summary

Table 4 lists the schemas defined in the preceding subsections. Underlined names are state schemas, others are operation schemas.

The table shows the schema inclusion hierarchy. Schemas are indented under the state schemas they include (for example the *SelectDisplay* operation and the *List* state both include the *Available* state). State schemas indented at the same level are mutually exclusive or independent of one another (*List* and *Table* are mutually exclusive while *List* and *Setup* are independent). Operation schemas are followed in parentheses by the operation schemas they include (so *DialogEdit* includes the *SelectItem* and *DialogOp* operations).

The table also shows transition involving certain state variables, especially *interaction* and *op*. Operation schema names are followed by their postconditions, so the postconditions of the *DialogOp* include the *Dialog* state and the postcondition of *SelectItem* include the *Editing* state. Only postconditions that indicate state changes are shown; $x' = x$ “no change” postconditions are not shown. Postconditions are not shown when they can be inferred from included operations, for example *DialogEdit* includes both *DialogOp* and *SelectItem*, so its postcondition includes both *Dialog* and *Editing*.

The table shows how the program can alternate between *Available* and *Engaged* states. Under *Available* the *DialogOp*, *MenuOp*, and *ConfirmOp* operations result in *Dialog*, *Menu* and *Confirm*, respectively, under *Engaged*. From there, the *Done* operations *Accept* and *Cancel* (etc.) return to *Available*.

The tables shows changes in *op*: *SelectDisplay* and *Done* set $op' = display'$, *SelectItem* sets op' to one of the *edit* operations. *GetChar*, the *GetArrow* operations and *Reprompt* do not change *op*. Other changes to *op'* are determined in the specific operations in the application (below), not these building blocks.

¹⁹In *EditOrInvalidSetting*, the two outputs *item!* and *value!* are not used in the *Invalid* case.

Unlocked

- Select
- Available

 - Op
 - SelectDisplay (Op), $op' = display' = input?$
 - SelectList (SelectDisplay), List'
 - SelectTable (SelectDisplay), Table'
 - ConfirmOp (Op), Confirm'
 - MenuOp (Op), Menu'
 - DialogOp (Op), Dialog'

- List

 - GetListArrow
 - SelectName (Select)

- Table

 - GetSettingArrow

- Setup

 - (Table)
 - SelectItem (Select), Editing', $op' = edit_setting \vee op' = edit_cal$
 - MenuEdit(SelectItem, MenuOp)
 - DialogEdit(SelectItem, DialogOp)

- Running
- Engaged

 - Done, Available', $op' = display$
 - Cancel (Done)
 - Confirm

 - AcceptConfirm (Select, Done)

 - Menu

 - GetMenuArrow
 - AcceptMenu (Select, Done)
 - (Editing)

 - MenuSettingC (AcceptMenu)

 - Dialog

 - GetChar
 - Accept (Done)
 - Reprompt
 - (Editing)

 - EditOrInvalidSetting (Accept or Reprompt)

Table 4: User interface building blocks

8.3 Therapy console operations

In this subsection we present the operations described in [2], in the order their constituent building blocks appear in Table 4.

Several building block operations require no further elaboration: *SelectDisplay*, *SelectTable*, *GetListArrow*, *GetSettingArrow*, *Cancel*, *GetMenuArrow*, *MenuSettingC*, *GetChar*, and *EditOrInvalidSetting* and are already complete. Others require further specialization in the following subsections.

8.3.1 Relation to Session state

A few *Console* operations read (but do not change) variables from the *Session* state (section 5). When the *Console* shows the patient or field list, its *nlist* state variable holds the patients or fields from the *Session* state. This is expressed by the *ConsoleSession* invariant:

<i>ConsoleSession</i> —
<i>Console</i>
<i>Session</i>
<i>display = select_patient</i> \Rightarrow <i>nlist = names</i>
<i>display = select_field</i> \Rightarrow <i>nlist = dom fields</i>

8.3.2 *Op* operations

Several operations are based only on *Op*. They are *Continue* operations because they do not involve any ongoing interaction, just a single keypress.

$$\text{SimpleOp} \hat{=} \text{Op} \wedge \text{Continue}$$

Experiment Mode (8.9.6, 190):

<i>ExptModeC</i> —
<i>SimpleOp</i>
$\Xi\text{Session}$
$\Delta\text{ConsoleSession}$
<i>Setup</i>
<i>input? = expt_mode</i>
<i>operator</i> \in <i>physicists</i>

Auto Setup (8.8.1, 181):

$auto_setup_display == \{field_summary, filter_wedge, leaf_collim, dose_intlk\}$

```
AutoSetupC _____  
SimpleOp  
ΞConsoleSession  
subsystem! : auto-setup-display  
  
Setup  
field ≠ no_field  
display ∈ auto-setup-display  
input? = auto-setup  
subsystem! = display
```

8.3.3 *SelectDisplay* operations

There are a few simple displays that provide no selections or interactive editing, **Field Summary** (8.9.7, 191) and the *help* display (not discussed in [2]):

$simple_display == \{field_summary, help\}$

8.3.4 *SelectList* operations

The specializations of *SelectList* are **Select Patient** (8.9.3, 184) and **Select Field** (8.9.4, 186). The latter operation only makes sense when a patient has been selected:

$list = \{select_patient, select_field\}$

```
SelectPatientList _____  
SelectList  
ΞSession  
ΔConsoleSession  
  
input? = select_patient  
nlist' = names
```

SelectFieldList

SelectList

\exists *Session*

Δ *ConsoleSession*

patient \neq *no_patient*

input? = *select_field*

nlist' = dom *fields*

8.3.5 *SelectTable* operations

The *table* displays are **Gantry/PSA** (8.9.8, 193), **Filter/Wedge** (8.9.9, 194), **Leaf Collimator** (8.9.10, 196), **Dosimetry-Therapy Interlocks** (8.9.11, 199) and **Calibration Factors** (8.9.13, 213):

table = {*gantry_psa*, *filter_wedge*, *leaf_collim*, *dose_intlk*, *dose_cal*}

The constant *table_items* tells which items on each table can be selected for editing or overriding (additional items may be displayed as well).

table_items = {*gantry_psa* \mapsto {*gantry*, *collim*, *turnt*}, *filter_wedge* \mapsto {*filter*, *wedge*, *w_rot*},
leaf_collim \mapsto *leaves*, *dose_intlk* \mapsto {*p_dose*, *p_time*},
dose_cal \mapsto {*pt_mode*, *press*, *temp*, *d_rate*, *t_fac*}}

It is necessary to separate *setting* and *dose_reg* items on different tables:

setting_table = {*gantry_psa*, *filter_wedge*, *leaf_collim*}

dose_reg_table = {*dose_intlk*, *dose_cal*}

Now that these constants are defined, the previously defined *SelectTable* operation requires no further specialization.

8.3.6 *ConfirmOp* operations

To begin **Cancel Run** (8.9.11, 209 – 210):

| *cancel_run_query* : CAPTION

<i>SelectCancelRun</i>	<hr/>
<i>ConfirmOp</i>	<hr/>
<i>Running</i>	
<i>input? = cancel_run</i>	
<i>op' = input?</i>	
<i>query! = cancel_run_query</i>	

The complementary *AcceptConfirm* operation is *CancelRunC* (below).

8.3.7 *MenuOp* operations

At this writing there are no simple *MenuOp* operations, only *MenuEdit* operations (under *Setup*, below).

8.3.8 *DialogOp* operations

<i>type_message_prompt, store_field_prompt : CAPTION</i>	
--	--

To begin **Write Log Message** (2.5.1, 17):

<i>TypeMessage</i>	<hr/>
<i>DialogOp</i>	<hr/>
<i>input? = log_message</i>	
<i>op' = input?</i>	
<i>prompt! = type_message_prompt</i>	

To begin **Store Field** (8.9.5, 189 – 188):

<i>EditField</i>	<hr/>
<i>DialogOp</i>	<hr/>
<i>Setup</i>	
<i>input? = store_field</i>	
<i>op' = input?</i>	
<i>prompt! = store_field_prompt</i>	

These two operations are completed by the complementary *Accept* operations, *WriteMessageC* and *StoreFieldC* (below).

8.3.9 Setup operations

Under *Setup*, there are *SelectName*, *SelectItem*, *MenuEdit* and *DialogEdit* operations.

Select Patient (8.9.3, 184 – 185):

```
SelectPatientC _____  
| SelectName  
|  
| Setup  
| display = select_patient  
| Continue  
|  
|_____
```

At this writing, *SelectPatient* is a *Continue* operation; the patient list remains on the screen²⁰.

SelectField (8.9.4, 186 – 189): There are three cases. The simplest case occurs during experiment mode, or when the chosen field has not yet been delivered today and the prescribed total dose and number of fractions has not yet been exceeded (8.9.4, 187).

```
NewFieldC _____  
| SelectName  
| ConsoleSession  
|  
| Setup  
| display = select_field  
|  
|_____
```

```
SelectSimpleFieldC _____  
| NewFieldC  
|  
| mode = experiment  
|  $\vee$  (counters name! dose = 0  $\wedge$   $\neg$  exceeded(fields name!, counters name!))  
| Continue  
|  
|_____
```

The more complicated cases arise in therapy mode when the operator must be warned of some unusual condition (8.9.4, 187-188). These are *DialogOp* operations. The name of the new field must be stored during the dialog. The operator may enter a preset dose or cancel the dialog (so no new field is selected).

²⁰We also considered establishing *display' = select_field* in *SelectPatientC*. It would not be difficult to adopt this alternative later.

Console1

ConsoleSession

new-field : FIELD

new-field \in dom *fields*

DoseDialogOp

NewFieldC

DialogOp

Δ *Console1*

op' = select-field

new-field' = name!

mode = therapy

There are two such cases. The first arises when the same field has already been delivered on the same day but the prescribed daily dose has not yet been reached; the remaining dose is offered as the default (8.9.4, 187):

sprintf : VALUE \rightarrow STRING

delivered-prompt : NAME \times VALUE \times VALUE \times VALUE \rightarrow CAPTION

SelectDeliveredField

DoseDialogOp

counters new-field' dose > 0

\neg *exceeded(fields new-field', counters new-field')*

(**let** *d == fields new-field' dose; c == counters new-field' dose* •

(**let** *default-dose == d - c* •

buffer' = sprintf default-dose \wedge

prompt! = delivered-prompt(new-field', d, c, default-dose))

The other case arises when the daily dose, the total dose or the number of fractions has been exceeded (8.9.4, 188). No default dose is provided.

exceeded-prompt : NAME \times ACCUMULATION \times ACCUMULATION \rightarrow CAPTION

SelectExceededField

DoseDialogOp

exceeded(fields new-field', counters new-field')

prompt! = exceeded-prompt(new-field', fields new-field', counters new-field')

buffer' = empty

The complete operation is composed of all these cases.

$$SelectFieldC \doteq SelectSimpleFieldC \vee SelectDeliveredField \vee SelectExceededField$$


After *SelectSimpleFieldC*, nothing more need be done. *SelectExceededField* and *SelectDeliveredField* are succeeded by the *SelectFieldOp* state, which is handled by the *SelectComplexFieldS* operation.



Override (8.4, 175; 8.8.1, 181; 8.8.2, 183) is also a *ConfirmOp* operation, enabled only when a field has been selected. The name of the item that the operator selected is echoed in the confirmation dialog. To begin **Override**:

$$override_table == \{filter_wedge, leaf_collim, gantry_psa, dose_intlk\}$$

$$| \quad override_query : CAPTION \rightarrow CAPTION$$



The complementary *AcceptConfirm* operations is *OverrideC* (below).

Edit: There are three cases. Calibration factors can be edited in experiment mode (8.9.13, 215), preset dose and time can be edited in both modes when a field is selected (8.9.11, 201–202), and other preset items can be edited in experiment mode when a field is selected (8.8.1, 180–181; 8.9.8, 194; 8.9.9, 196; 8.9.10, 198).

$$\begin{aligned} cal_table &== \{dose_cal\} \\ dose_table &== \{dose_intlk\} \\ preset_table &== \{filter_wedge, leaf_collim\} \end{aligned}$$

$$CalTable \hat{=} [ConsoleSession \mid mode = experiment \wedge display \in cal_table]$$

The precondition $field \neq no_field$ occurs elsewhere so it is convenient to collect two cases together.

<i>SettingTable</i>	_____
<i>ConsoleSession</i>	
	$display \in dose_table \vee (mode = experiment \wedge display \in preset_table)$

Finally

$$\begin{aligned} SelectCalMenu &\hat{=} CalTable \wedge MenuEdit \\ SelectCalDialog &\hat{=} CalTable \wedge DialogEdit \\ SelectSettingMenu &\hat{=} SettingTable \wedge MenuEdit \\ SelectSettingDialog &\hat{=} SettingTable \wedge DialogEdit \end{aligned}$$

The complementary operations are $EditSettingC$ and $MenuSettingC$ (section 8.2, above).

8.3.10 Cancel operations

There is a special *cancel* operation for the login process (below) so we have to strengthen the preconditions on *Cancel*:

$$\begin{aligned} LoggedIn &\hat{=} [Console \mid op \notin \{login, password\}] \\ CancelOp &\hat{=} LoggedIn \wedge Cancel \end{aligned}$$

8.3.11 AcceptConfirm operations

To complete **Override** (8.4, 175; 8.8.1, 181; 8.8.2, 183):

<i>OverrideC</i>	_____
<i>AcceptConfirm</i>	
<i>item! : ITEM</i>	
	$op = override_cmd$
	$item! = item$

To complete **Cancel Run** (8.9.11, 209 – 210):

$$CancelRunC \hat{=} [AcceptConfirm \mid op = cancel_run]$$

8.3.12 Accept operations

To complete **Write Log Message** (2.5.1, 17). The *smessage* function turns a string into a log message by prepending the timestamp and other information.

| *log-msg* : STRING —> MESSAGE

WriteMessageC —————

Accept

message! : MESSAGE

op = *log-message*

message! = *log-msg buffer*

To complete **Store Field** (8.9.5, 189 – 188)²¹.

| *sname* : STRING —> NAME
store-msg : NAME —> MESSAGE

StoreFieldC —————

Accept

field! : NAME

message! : MESSAGE

op = *store-field*

field! = *sname buffer*

message! = *store-msg field!*

The complex variants of **Select Field** (8.9.4, 187–188) are handled by *SelectComplexFieldC*, which is similar to *EditSettingC*:

SelectComplexFieldC —————

Δ Console1

Accept

field! : FIELD

dose! : VALUE

SelectFieldOp

field! = *new-field*

(let *d* == *sval buffer* • *d* ∈ *valid dose* ∧ *dose!* = *d*)

²¹The *message!* output from *StoreFieldC* is not mentioned in [2]; here we correct the omission.

We have to make the operation total


$$InvalidDose \triangleq [Reprompt; \Delta Console1 \mid SelectFieldOp \wedge sval\ buffer \notin valid\ dose]$$

$$ComplexOrInvalidField \triangleq SelectComplexFieldC \vee InvalidDose$$


8.3.13 Logout and login

The **Login** process (Fig. 2.6, 19) can be seen as editing the value of *operator*. Messages are logged at logout and login. **Logout** (2.5.2, 17 – 18; 8.9.2, 184) is similar to *SelectDisplay* and *DialogOp*; it reads *Session* for the operator's ID in the logout message.

$$| \quad o_msg, lo_msg : OPERATOR \rightarrow MESSAGE$$



A successful **Login** (2.5.2, 17 – 20; 8.9.1, 183) occurs when a user enters a valid operator identification. The process of logging back in is broken into two steps. In the first step, the user types their username into the buffer. When the user types a terminator the username is saved in another buffer; the *Console1* state is *Console* with this buffer added²². The console remains logged out after this step. This *EnterUsername* operation is similar to *Accept* and *Continue*.



²²The password is also typed into *buffer* so the *GetChar* operation can be used.

EnterUsername

$\Delta Console2$
 $\Delta Dialog$
 $EventUnlocked$

$op = login$
 $input? \in terminator$
 $username' = buffer$
 $buffer' = empty$
 $op' = password$
 $display' = display$

In the second step, the user enters their password. If their username, password pair is found in operator database, the user is logged in, the help screen is displayed, and the console becomes available. This *LoginC* operation is similar to *Accept* and *SelectDisplay*.

The operator authorization file is modelled by Z global constants *operators* and *physicists* (physicists are authorized to use the equipment in its experiment mode, see 8.2, 170). Values of *OPERATOR* include the operator's password in addition to the operator's printed name.

USERNAME == STRING; PASSWORD == STRING

| *soper : (USERNAME × PASSWORD) → OPERATOR*

LoginC

$\Delta Console2$
 $EventUnlocked$
 $operator! : OPERATOR$
 $message! : MESSAGE$

$op = password$
 $soper(username, buffer) \in operators$
 $input? \in terminator$
 $display' = help$
 $op' = display$
 $operator! = soper(username, buffer)$
 $message! = o_msg operator!$
 $Available'$

If the username, password pair is not found in the authorization file, the console remains logged out.

<i>Unauthorized</i>	_____
$\Delta Console2$	
<i>Reprompt</i>	
	$op = password$ $soper(username, buffer) \notin operators$ $username' = username$

$$LoginOrUnauthorized \cong LoginC \vee Unauthorized$$

Users may cancel a login attempt while entering their username or password. The login process begins anew.

<i>CancelUsername</i>	_____
$\Delta Console2$	
$\Delta Dialog$	
<i>EventUnlocked</i>	
	$op \in \{login, password\}$ $input? = cancel$ $op' = login$ $buffer' = empty$ $username' = empty$ $display' = display$

8.3.14 Other operations

Several operations not described in [2] are included for development purposes²³. They are always enabled.

$$\begin{aligned} Refresh &\cong [Event \mid input? = refresh] \\ Shutdown &\cong [Event \mid input? = shutdown] \end{aligned}$$

This concludes our presentation of the states and operations in the user interface.

8.3.15 Summary

Table 5 shows all of the top-level operations that the user can invoke, in almost the same format as the building block operations in Table 4. Some building block operations also

²³*Shutdown* will not be included in production versions.

appear in Table 5; new operations are followed in parentheses by the names of the building block operations they use.

8.4 Implementation

In [4] we describe how to implement a user interface specified in our style. The method requires that each operation be expressed as a conjunction of three separate schemas: a state schema for the preconditions involving the state variables, a state schema for the preconditions involving the input variable, and an operation schema. From the operation schemas we have already defined, we factor out the following schemas for states and inputs:

$$\begin{aligned}
Physicist &\hat{=} [ConsoleSession \mid operator \in physicists] \\
PatientList &\hat{=} [Console \mid display = select_patient \wedge nlist \neq \emptyset] \\
PatientSelected &\hat{=} [ConsoleSession \mid patient \neq no_patient] \\
FieldList &\hat{=} [Console \mid display = select_field \wedge nlist \neq \emptyset] \\
FieldSelected &\hat{=} [ConsoleSession \mid field \neq no_field] \\
AutoSetupDisplay &\hat{=} [Console \mid display \in auto_setup_display] \\
OverrideTable &\hat{=} [Console \mid display \in override_table] \\
MenuItem &\hat{=} [Console \mid item \in selection] \\
DialogItem &\hat{=} [Console \mid item \notin selection] \\
LoggedOut &\hat{=} [Console \mid op \in \{login, password\}] \\
OverrideOp &\hat{=} [Console \mid op = override_cmd] \\
CancelRunOp &\hat{=} [Console \mid op = cancel_run] \\
LogMessageOp &\hat{=} [Console \mid op = log_message] \\
StoreFieldOp &\hat{=} [Console \mid op = store_field] \\
UsernameOp &\hat{=} [Console \mid op = login] \\
PasswordOp &\hat{=} [Console \mid op = password] \\
\\
Input &\hat{=} [input? : INPUT] \\
DisplayKey &\hat{=} [Input \mid input? \in simple_display] \\
PatientKey &\hat{=} [Input \mid input? = select_patient] \\
FieldKey &\hat{=} [Input \mid input? = select_patient]
\end{aligned}$$

$$\begin{aligned}
TableKey &\hat{=} [Input \mid input? \in table] \\
MessageKey &\hat{=} [Input \mid input? = log_message] \\
VArrowKey &\hat{=} [Input \mid input? \in v_arrow] \\
SelectKey &\hat{=} [Input \mid input? = select] \\
ArrowKey &\hat{=} [Input \mid input? \in arrow] \\
ExptModeKey &\hat{=} [Input \mid input? = expt_mode] \\
AutoSetupKey &\hat{=} [Input \mid input? = auto_setup] \\
StoreFieldKey &\hat{=} [Input \mid input? = store_field] \\
LoginKey &\hat{=} [Input \mid input? = login] \\
OverrideKey &\hat{=} [Input \mid input? = override_cmd] \\
CancelRunKey &\hat{=} [Input \mid input? = cancel_run] \\
CancelKey &\hat{=} [Input \mid input? = cancel] \\
CharKey &\hat{=} [Input \mid input? \in CHAR] \\
TerminatorKey &\hat{=} [Input \mid input? \in terminator] \\
RefreshKey &\hat{=} [Input \mid input? = refresh] \\
ShutdownKey &\hat{=} [Input \mid input? = shutdown]
\end{aligned}$$

Table 6 (essentially Table 5 reformatted) expresses the entire user interface in the format required by our implementation method [4]²⁴. The table represents *ConsoleOp*: all the top level operations from Table 5 combined into a single operation.

ConsoleOp $\hat{=} SelectDisplay \vee SelectPatientList \vee SelectFieldList \vee \dots \vee IgnoreOthers$

IgnoreOthers is the default do-nothing operation.

IgnoreOthers $\hat{=} [Ignore \mid \dots]$

IgnoreOthers ensures that *ConsoleOp* is total; its precondition is the negation of the disjunction of the preconditions of all the other operations. We do not define this precondition explicitly; we code the implementation so control reaches *IgnoreOthers* when no other operations are enabled.

ConsoleOp defines a state transition system where each disjunct defines a single transition. Table 6 is the state transition table. There is an entry (row) in the table for each top-level operation schema. The first column names the state precondition, the second column names

²⁴The LATEX source for Table 6 and code skeletons for the implementation (in C) are generated automatically (by sed and awk scripts) from the same text file.

Refresh
Shutdown
Unlocked
Available
 SelectDisplay (Op)
 SelectList (SelectList)
 SelectTable (SelectDisplay)
 TypeMessage (DialogOp)
List
 GetListArrow
Table
 GetSettingArrow
PatientSelected
 SelectList (SelectList)
Setup
 Logout (similar to SelectDisplay, DialogOp)
 ExptModeC (Op)
(List)
 SelectPatientC (SelectName)
 SelectFieldC (SelectName, DialogOp)
(PatientSelected)
 EditField (DialogOp)
(CalTable)
 SelectMenuItem (MenuEdit)
 SelectDialogItem (DialogEdit)
(FieldSelected)
 AutoSetupC (Op)
 SelectOverride (SelectItem, ConfirmOp)
(SettingTable)
 SelectMenuItem (MenuEdit)
 SelectDialogItem (DialogEdit)
Running
 SelectCancelRun (ConfirmOp)
Engaged
(LoggedIn)
 CancelOp (Cancel)
Confirm
 OverrideC (AcceptConfirm)
 CancelRunC (AcceptConfirm)
Menu
 GetMenuArrow
(Editing)
 MenuSettingC (AcceptMenu)
Dialog
 GetChar
 WriteMessageC (Accept)
 ComplexOrInvalidField (Accept or Reprompt)
 StoreFieldC (Accept)
(Editing)
 EditOrInvalidSetting (Accept or Reprompt)
(LoggedOut)
 CancelUsername (similar to Cancel)
 EnterUsername (similar to Accept, Continue)
 LoginOrUnauthorized (similar to Accept and SelectDisplay, or Reprompt)

Table 5: Therapy operations

State precondition	Input precondition	Operation
0 Z_True	RefreshKey	Refresh
0 Z_True	ShutdownKey	Shutdown
0 Unlocked	NoKey	NoOp
1 Available	DisplayKey	SelectDisplay
1 .	PatientKey	SelectPatientList
1 .	TableKey	SelectTable
1 .	MessageKey	TypeMessage
2 List	VArrowKey	GetListArrow
2 Table	ArrowKey	GetSettingArrow
2 PatientSelected	FieldKey	SelectFieldList
2 Setup	LoginKey	Logout
3 Physicist	ExptModeKey	ExptModeC
3 PatientList	SelectKey	SelectPatientC
3 FieldList	SelectKey	SelectFieldC
3 PatientSelected	StoreFieldKey	EditField
3 CalTable	NoKey	NoOp
4 MenuItem	SelectKey	SelectMenuItem
4 DialogItem	SelectKey	SelectDialogItem
3 FieldSelected	NoKey	NoOp
4 AutoSetupDisplay	AutoSetupKey	AutoSetupC
4 OverrideTable	OverrideKey	SelectOverride
4 SettingTable	NoKey	NoOp
5 MenuItem	SelectKey	SelectMenuItem
5 DialogItem	SelectKey	SelectDialogItem
2 Running	CancelRunKey	SelectCancelRun
1 Engaged	NoKey	NoOp
2 LoggedIn	CancelKey	CancelOp
2 Confirm	NoKey	NoOp
4 OverrideOp	SelectKey	OverrideC
4 CancelRunOp	SelectKey	CancelRunC
2 Menu	VArrowKey	GetMenuArrow
3 Editing	SelectKey	MenuSettingC
2 Dialog	CharKey	GetChar
3 LogMessageOp	TerminatorKey	WriteMessageC
3 SelectFieldOp	TerminatorKey	ComplexOrInvalidField
3 StoreFieldOp	TerminatorKey	StoreFieldC
3 Editing	TerminatorKey	EditOrInvalidSetting
3 LoggedOut	CancelKey	CancelUsername
3 UserNameOp	TerminatorKey	EnterUsername
3 PasswordOp	TerminatorKey	LoginOrUnauthorized

Table 6: Therapy console state transition table

the input precondition, and the last column names the operation schema itself. In this table the operations appear in the same order as they do in Table 5.

In our table, sequence order and nesting level (indicated by indentation and the number in the first column) represent the nesting of states that is expressed in Z by schema inclusion. A greater nesting level indicates that a table entry is a *substate* of preceding entries at lesser nesting levels. *The full state precondition of a substate is formed by conjoining the state preconditions of the preceding entries at lesser nesting levels.* For example, the full state precondition for the *SelectPatientC* operation is *Unlocked* \wedge *Available* \wedge *Setup* \wedge *PatientList*. *When the first column in a row is blank, the substate is the same as the last preceding nonblank column at the same nesting level:* the precondition for the *Logout* operation is *Unlocked* \wedge *Available* \wedge *Setup*. Each line also applies to any included substates, so *DisplayKey* elicits the *SelectDisplay* operation in the *Available* state, and also in its substates *Setup* and *Running*, and in its sub-substates *PatientList* and *FieldList* etc.

The table requires these place holders.

$$\text{True} \hat{=} \text{Console}$$

$$\text{NoOp} \hat{=} \exists \text{Console}$$

$$\text{NoKey} \hat{=} [\text{Input} \mid \text{false}]$$

9 Combining the subsystems

In this section we combine related operations from the *Console*, *Session* and *Field* subsystems. In cases where no data is transferred between subsystems, we can simply conjoin the separate operations:

$$\text{ExptMode} \hat{=} \text{ExptModeC} \wedge \text{ExptModeS} \wedge \text{ExptModeF}$$

Here the conjunction just expresses that the named operations in all three subsystems are triggered by the *expt_mode* input at the console. In this report both *ExptModeC* and *ExptModeS* contain the precondition *operator* \in *physicists*, but this predicate only needs to appear once; we include it in both schemas for clarity²⁵.

In other cases, data is transferred. For example, in the *Override* operation, the *item!* output from the *Console* subsystem is consumed by the *item?* input in the *Field* subsystem (the *Session* subsystem does not participate in this operation). We wish to express

²⁵In our implementation, we observe the convention that preconditions of combined operations are always tested in the *Console* operations. For example *operator* \in *physicists* is tested by code in *zconsole.c*, not *zsession.c*.

<i>Override</i>	_____
<i>OverrideC</i>	
<i>OverrideF</i>	
<i>item! = item?</i>	

This can be expressed more concisely using the Z pipe operator *pipe*, which has the effect of connecting corresponding input and output variables ([7], p. 78)²⁶.

$$\textit{Override} \doteq \textit{OverrideC} \gg \textit{OverrideF}$$

In *Login*, the new operator name *operator!/operator?* is the piped variable.

$$\textit{Login} \doteq \textit{LoginC} \gg \textit{LoginS}$$

In *EditSetting*, both the item name *item!/item?* and the new setting value *value!/value?* are piped:

$$\textit{EditSetting} \doteq \textit{EditSettingC} \gg \textit{EditSettingF}$$

Sometimes the output from the *Console* subsystem is piped to inputs in both the *Session* and *Field* subsystems. In *StoreField*, the new field name *field!/field?* is the piped variable:

$$\textit{StoreField} \doteq \textit{StoreFieldC} \gg (\textit{StoreFieldS} \wedge \textit{StoreFieldF})$$

Here the conjunction *StoreFieldS* \wedge *StoreFieldF* ensures that *prescribed'* in *fields' = fields* \cup $\{\textit{field}' \mapsto \textit{prescribed}'\}$ from *StoreFieldS* is the same as *prescribed'* in *prescribed' = prescribed* \oplus *measured* ... from *StoreFieldF*.

In *SelectPatient*, the new patient name *patient!/patient?* is the piped variable (*patient?* does not appear in *SelectPatientF*).

$$\textit{SelectPatient} \doteq \textit{SelectPatientC} \gg (\textit{SelectPatientS} \wedge \textit{SelectPatientF})$$

There are two variations of *SelectField*. In both variants *field!/field?* is piped from the *SelectFieldC* variant to *SelectFieldS*. In the complex variant, *dose!/dose?* is piped from *SelectComplexFieldS* to *SelectComplexFieldF*.

$$\textit{SelectSimpleField} \doteq \textit{SelectSimpleFieldC} \gg (\textit{SelectFieldS} \wedge \textit{SelectSimpleFieldF})$$

$$\textit{SelectComplexField} \doteq \textit{SelectComplexFieldC} \gg (\textit{SelectFieldS} \wedge \textit{SelectComplexFieldF})$$

²⁶The Z pipe operator also hides the piped variables.

References

- [1] Jonathan Jacky, Ruedi Risler, Ira Kalet, and Peter Wootton. Clinical neutron therapy system, control system specification, Part I: System overview and hardware organization. Technical Report 90-12-01, Radiation Oncology Department, University of Washington, Seattle, WA, December 1990.
- [2] Jonathan Jacky, Ruedi Risler, Ira Kalet, Peter Wootton, and Stan Brossard. Clinical neutron therapy system, control system specification, Part II: User operations. Technical Report 92-05-01, Radiation Oncology Department, University of Washington, Seattle, WA, May 1992.
- [3] Jonathan Jacky and Jonathan Unger. Formal specification of control software for a radiation therapy machine. Technical Report 94-07-01, Radiation Oncology Department, University of Washington, Seattle, WA, July 1994.
- [4] Jonathan Jacky and Jonathan Unger. From Z to code: A graphical user interface for a radiation therapy machine. In J. P. Bowen and M. G. Hinchey, editors, *ZUM '95: The Z Formal Specification Notation*, pages 315 – 333. Ninth International Conference of Z Users, Springer-Verlag, 1995. Lecture Notes in Computer Science 967.
- [5] Adrian Nye. *Xlib Programming Manual*. O'Reilly and Associates, Inc., Sebastopol, CA, 1988.
- [6] J. M. Spivey. *The fuzz Manual*. J. M. Spivey Computing Science Consultancy, Oxford, second edition, July 1992.
- [7] J. M. Spivey. *The Z Notation: A Reference Manual*. Prentice-Hall, New York, second edition, 1992.

A Glossary of items

A.1 Settings

These are the elements of *setting*:

collim Collimator rotation angle (8.9.8, Fig. 8.5, 193 – 194).

dose Dose per fraction. The *prescribed* value is the dose prescribed to be delivered from the current field in a single fraction. It is read from the prescription file, and appears in the MU column of the field list display (Fig 8.3, 186) and the PRESCR column of the dosimetry display (Fig. 8.8, 199 etc.). The same *prescribed* values appear in both the A and B rows of the dosimetry display. The *measured* value is the dose accumulated in DMC channel A since it was last reset, in monitor units. It appears in the ACCUM column in the dosimetry display. The *accumulated* value is the total dose accumulated in all treatment runs since the beginning of the day, in monitor units (8.9.4, 187 bottom half – 188 top half). See also the *dose_reg* element *p_dose*, below.

doseB Dose accumulated in DMC channel B since it was last reset, in monitor units. This backup to channel A is used internally by the DMC but its only role in our control program is to be displayed in the ACCUM column in the dosimetry display.

dose_tot Total dose intended to be delivered from the current field over the entire course of treatment, in monitor units. The *prescribed* value is from the prescription file, and appears in the Total column of the field list display (Fig 8.3, 186). The *accumulated* value is the dose actually delivered to date, in monitor units (8.9.4, Fig 8.3, 186; 187 bottom half – 188 top half), and appears in the last To date column on the field list display (Fig 8.3, 186).

filter Flattening filter selection (no_filter, small field, large field) (8.9.9, Fig. 8.6, 194 – 196).

gantry Gantry rotation rotation angle (8.9.8, Fig. 8.5, 193 – 194).

height Couch height position. Table positions are not included in determining *ready* status, but may be stored in prescription file (8.9.8, Fig. 8.5, 193 – 194).

lat Couch lateral position (8.9.8, Fig. 8.5, 193 – 194).

leaf0, leaf39 Collimator leaf positions (8.9.10, Fig. 8.7, 196 – 198). There are actually forty leaves, not just two, but for brevity we omit *leaf1..leaf38* from the formal description. Our *leaf0* has the same *valid* values as *leaf1..leaf19*, and *leaf39* has the same *valid* values as *leaf20..leaf38*.

longit Couch longitudinal position (8.9.8, Fig. 8.5, 193 – 194).

nfrac Intended total number of fractions for this field, over the entire course of treatment (8.9.4, 188). The *prescribed* value appears in the first Fractions column on the field list display (Fig 8.3, 186). The *accumulated* value is the number of fractions accumulated for this field to date (8.9.4, 187 – 188), and appears in the first To date column on the field list display (Fig 8.3, 186).

top Couch top swivel rotation (8.9.8, Fig. 8.5, 193 – 194).

turnt Turntable rotation (8.9.8, Fig. 8.5, 193 – 194).

w_rot Wedge rotation (8.9.9, Fig. 8.6, 194 – 199).

wedge Wedge selection (8.9.9, Fig. 8.6, 194 – 199).

A.2 Registers

These are the elements of *dose-reg*:

calvolt1 DMC standard calibration voltage. The *calibrated* value is read from a file, while the *computed* value is computed from the *calibrated* value by adjusting by a pressure-temperature correction factor (8.9.13, 213 – 214). The *computed* value is actually loaded into the DMC before each run.

calvolt2 DMC standard calibration voltage, etc.

d_rate Nominal dose rate, “dose rate of the day,” used to calculate treatment backup time, *p_time* (8.9.11, 200, last paragraph, 202, second paragraph; 8.9.13, 213, first two paragraphs after bullets). The *calibrated* value is read from a file; the *computed* value is initialized to the same value but may be edited by the operator.

e_time Elapsed time. The *calibrated e_time* is the beam-on time elapsed since the DMC was last reset, and appears in ACCUM column in dosimetry display (8.9.11, Fig. 8.8, 199 etc.)

p_dose Preset dose loaded into the DMC at the beginning of each treatment run. The *computed* value is computed by the control program when the field is selected and is usually set equal to the *prescribed dose*, except (for example) after an interrupted treatment run. Subsequently it may be edited by the operator (8.9.4, 187 - 188; 8.9.11, 201 – 202). It appears in the PRESET column of the dosimetry display. The same *computed* values appear in both the A and B rows of the dosimetry display. See also the *setting* elements *dose* and *doseB*.

p_time Preset time, the treatment backup time. The *calibrated p_time* is initially calculated from preset dose *p_dose*, dose rate *d_rate*, and time factor *t_fac* (8.9.11, 202; 8.9.13, 213); it appears in PRESCR column in dosimetry display (Fig 8.8, 199 etc.). The *computed p_time* is initially set equal to *calibrated p_time* but may be edited by the operator (8.9.11, 202; 8.9.13, 213). The *computed p_time* is actually loaded into the DMC before each run.

press Barometric pressure used to compute pressure/temperature correction factor. The *calibrated* value is measured continuously, while the *computed* value is entered by the operator (8.9.13, 213 – 214).

pt_factor Pressure/temperature correction factor used to compute *computed calvolt1* and *computed calvolt2* (8.9.13, 214). The *calibrated* value is computed from *calibrated press* and *calibrated tempo* measured from sensors, while the *computed* value is computed from *computed press* and *computed tempo* entered by the operator.

pt_mode Pressure/temperator correction factor selection (either automatic or manual) (8.9.13, 214). In *automatic* mode, *computed calvolt1* and *computed calvolt2* are calculated from *calibrated pt_factor* derived from sensor readings, while in *manual* mode they are calculated from the *computed pt_factor* derived from values entered by the operator.

t_fac Treatment time factor (8.9.11, 200, last paragraph, 202, second paragraph; 8.9.13, 213, first two paragraphs after bullets). The *calibrated* value is read from a file; the *computed* value is initialized to the same value but may be edited by the operator.

temp Temperature used to compute pressure/temperature correction factor. The *calibrated* value is measured by sensors, while the *computed* value is entered by the operator (8.9.13, 213 – 214).

B Groups of items

cal_const Calibration constants stored in files: *d_rate*, *t_fac*, *calvolt1* and *calvolt2* (8.9.13, 213).

counter Items whose actual values increase during a treatment run (such as *dose*) or over the entire course of treatment (such as the number of fractions *nfrac* and total dose *dose_tot*). Domain of *accumulated* in *Field*. Counters in *prescr* can only be *ready* when their *accumulated* values are less than their *prescribed* values. The only three counters are *dose*, *n* and *dose_tot*.

dose_reg Calibration factors and other items related to the dosimetry system which are not stored prescription files. Some are stored in calibration files, others are computed or entered by the operator. Domain of *calibrated* and *computed* in *Field*. They are *pt_mode*, *pt_factor*, *press*, *temp*, *d_rate*, *t_fac*, *calvolt1*, *calvolt2*, *p_dose*, *p_time*, and *e_time*.

motion Settings that represent internal or external motions (for example, wedge rotation and gantry rotation, respectively). Each motion is either enabled or disabled (8.5, bottom 176 – top 177). The domain of *drive* in *Intlk*. They are all the *leaves* including *leaf0* and *leaf39*, the filter settings *wedge*, *w_rot*, *filter*, the rotations *gantry*, *collim*, *turnt*, and the linear table motions *height*, *lat* and *long*.

prescrip Settings whose values are read from the prescription file for the selected patient and field. Subsequently, some may be edited by the operator. The domain of *prescribed* in *Field*. They are all the motions and all the counters (8.2, 171, third bullet).

prescr Settings whose prescribed values are checked against measured values in therapy mode. The domain of *status* in the *Intlk* schema. Same as *prescrip* except it excludes the three linear table motions, *height*, *lat* and *long*.

preset Settings whose values are read from the file of presets for the selected experiment field. They include only the leaves and the filter settings, omitting the external motions and the counters.

scale Items that vary continuously over some range, such as leaf position or gantry rotation. Scales in *prescr* are *ready* when their *measured* values lie within tolerance of their *prescribed* values. The domain of the global function *tol*. They include all the motions except the filter settings, and all the registers except the pressure-temperature factor correction mode.

selection Items that only take on discrete values, such as flattening filter selection. Selections are *ready* only when their *measured* values are exactly equal to their *prescribed* values. They are the three filter selections *filter*, *wedge*, and *w_rot* and the pressure-temperature factor correction mode *pt_mode*.

sensor Settings whose values are measured by sensors. Domain of function *measured* in *Field* schema. They are all the settings except the two counters *nfrac* and *dose_tot*.

setting All the items stored in the prescription database, with *doseB* and *top* as well.

This information is summarized in Table 7.

<i>ITEM</i>	<i>setting</i>	<i>d_reg.</i>	<i>cal.</i>	<i>prescrip</i>	<i>prescr</i>	<i>preset</i>	<i>motion</i>	<i>scale</i>	<i>sel.</i>	<i>counter</i>
<i>nfrac</i>	•			•	•					•
<i>dose_tot</i>	•			•	•					•
<i>dose</i>	•			•	•					•
<i>wedge</i>	•			•	•	•	•		•	
<i>w_rot</i>	•			•	•	•	•		•	
<i>filter</i>	•			•	•	•	•		•	
<i>leaf0</i>	•			•	•	•	•	•		
<i>leaf39</i>	•			•	•	•	•	•		
<i>gantry</i>	•			•	•		•	•		
<i>collim</i>	•			•	•		•	•		
<i>turnt</i>	•			•	•		•	•		
<i>lat</i>	•			•			•	•		
<i>longit</i>	•			•			•	•		
<i>height</i>	•			•			•	•		
<i>doseB</i>	•									
<i>top</i>	•									
<i>pt_mode</i>		•								•
<i>pt_factor</i>		•								
<i>press</i>		•								
<i>temp</i>		•								
<i>d_rate</i>		•	•							
<i>t_fac</i>		•	•							
<i>calvolt1</i>		•	•							
<i>calvolt2</i>		•	•							
<i>p_dose</i>		•								
<i>p_time</i>		•								
<i>e_time</i>		•								

Table 7: Groups of items

C Types and constants

This appendix collects together the types and constants that define the system configuration.

C.1 Settings and registers

From section 3.1.

$$\begin{aligned} ITEM ::= & nfrac \mid dose_tot \mid dose \mid wedge \mid w_rot \mid filter \mid leaf0 \mid leaf39 \mid \\ & gantry \mid collim \mid turnt \mid lat \mid longit \mid height \mid doseB \mid top \mid \\ & pt_mode \mid pt_factor \mid press \mid temp \mid d_rate \mid t_fac \mid \\ & calvolt1 \mid calvolt2 \mid p_dose \mid p_time \mid e_time \end{aligned}$$

$setting, dose_reg : \mathbb{P} ITEM$
$\langle setting, dose_reg \rangle$ partition $ITEM$
$dose_reg = \{pt_mode, pt_factor, press, temp, d_rate, t_fac,$
$calvolt1, calvolt2, p_dose, p_time, e_time\}$

$scale, selection, counter : \mathbb{P} ITEM$
$\langle selection, scale, counter \rangle$ partition $ITEM$
$counter = \{nfrac, dose_tot, dose\}$
$selection = \{wedge, w_rot, filter, pt_mode\}$

$$\begin{aligned} leaves &== \{leaf0, leaf39\} \\ preset &== leaves \cup \{wedge, w_rot, filter\} \\ motion &== preset \cup \{gantry, collim, turnt, lat, longit, height\} \\ prescrip &== motion \cup counter \\ prescr &== prescrip \setminus \{lat, longit, height\} \\ sensor &== setting \setminus \{nfrac, dose_tot\} \\ cal_const &== \{d_rate, t_fac, calvolt1, calvolt2\} \end{aligned}$$

C.2 Values

From section 3.2.

$$VALUE == \mathbb{Z}$$

$$\begin{array}{l} \boxed{\begin{array}{l} blank : VALUE \\ tol : scale \rightarrow VALUE \\ valid : ITEM \rightarrow \mathbb{P} VALUE \end{array}} \\ \hline \forall s : ITEM \bullet blank \notin valid s \end{array}$$

C.3 Prescription database

From section 4.

$$[NAME]$$

$$PATIENT == NAME; FIELD == NAME$$

$$\boxed{\begin{array}{l} no_name : NAME \end{array}}$$

$$no_patient == no_name; no_field == no_name$$

$$\boxed{\begin{array}{l} studies, patients : \mathbb{P} PATIENT \end{array}}$$

$$\boxed{\begin{array}{l} no_patient \notin studies \wedge no_patient \notin patients \end{array}}$$

$$ACCUMULATION == counter \rightarrow VALUE$$

$$PRESCRIPTION == prescrip \rightarrow VALUE$$

$$\boxed{\begin{array}{l} Preset : studies \rightarrow (FIELD \leftrightarrow PRESCRIPTION) \end{array}}$$

$$\boxed{\begin{array}{l} Prescribed : patients \rightarrow (FIELD \leftrightarrow PRESCRIPTION) \end{array}}$$

$$\boxed{\begin{array}{l} Accumulated : patients \rightarrow (FIELD \leftrightarrow ACCUMULATION) \end{array}}$$

$$\boxed{\begin{array}{l} \forall s : studies \bullet no_field \notin \text{dom}(Preset s) \end{array}}$$

$$\boxed{\begin{array}{l} \forall p : patients \bullet no_field \notin \text{dom}(Prescribed p) \wedge \text{dom}(Prescribed p) = \text{dom}(Accumulated p) \end{array}}$$

$$\boxed{\begin{array}{l} exceeded_ : \mathbb{P}(ACCUMULATION \times PRESCRIPTION) \end{array}}$$

$$\boxed{\begin{array}{l} \forall counters : FIELD \rightarrow ACCUMULATION; fields : FIELD \rightarrow PRESCRIPTION \bullet \end{array}}$$

$$\boxed{\begin{array}{l} exceeded(counters, fields) \Leftrightarrow (\exists c : counter \bullet counters c \geq fields c) \end{array}}$$

C.3.1 Operators

From section 4.1.

[*OPERATOR*]

$$\frac{\text{operators}, \text{physicists} : \mathbb{P} \text{OPERATOR}}{\text{physicists} \subseteq \text{operators}}$$

C.4 Session

From section 5.

MODE ::= therapy | experiment

C.5 Field

From section 6.

| *automatic, manual : VALUE*

| *cal_factor : cal_const → VALUE*

PRESSURE == VALUE; TEMPERATURE == VALUE

DOSE == VALUE; RATE == VALUE; FACTOR == VALUE; TIME == VALUE

| *t_backup : (DOSE × RATE × FACTOR) → TIME*

| *pt_formula : (PRESSURE × TEMPERATURE) → FACTOR*

C.6 User interface

From section 8.

$[CAPTION, MESSAGE]$

$alert : CAPTION$
 $ocaption : OP \rightarrow CAPTION$

$RUN ::= setup | running$

$KEYSWITCH ::= locked | unlocked$

$INTERACTION ::= available | dialog | menu | confirm$

$INPUT ::= filter_wedge | leaf_collim | dose_intlk | gantry_psa | dose_cal |$
 $startup | help | messages | select_patient | select_field | field_summary |$
 $login | edit_setting | edit_dose_reg | log_message | store_field | override_cmd |$
 $cancel_run | password | auto_setup | expt_mode | cancel | refresh | shutdown |$
 $select | ret | character | backspace | delete_key |$
 $left_arrow | right_arrow | up_arrow | down_arrow | ignored$

$OP : \mathbb{P} INPUT$

$OP = \{filter_wedge, leaf_collim, dose_intlk, gantry_psa, dose_cal,$
 $startup, help, messages, select_patient, select_field, field_summary,$
 $login, edit_setting, edit_dose_reg, log_message, store_field, override_cmd, cancel_run,$
 $password, auto_setup, expt_mode, cancel, refresh, shutdown, select\}$

$DISPLAY : \mathbb{P} OP$

$DISPLAY = \{filter_wedge, leaf_collim, dose_intlk, gantry_psa, dose_cal,$
 $startup, help, messages, select_patient, select_field,$
 $field_summary, login\}$

| $list, table : \mathbb{P} DISPLAY$

$default_item : table \rightarrow ITEM$
$table_items : table \rightarrow \mathbb{P} ITEM$
$setting_table, dose_reg_table : \mathbb{P} table$

$\forall d : table \bullet default_item d \in table_items d$
$\forall d : setting_table \bullet table_items d \subseteq setting$
$\forall d : dose_reg_table \bullet table_items d \subseteq dose_reg$

$v_arrow == \{up_arrow, down_arrow\}$

$arrow == \{right_arrow, left_arrow\} \cup v_arrow$

$asetting : (arrow \times ITEM \times table) \rightarrow ITEM$
$aname : (v_arrow \times NAME \times \mathbb{P}_1 NAME) \rightarrow NAME$

$\forall a : arrow; s : ITEM; d : table \bullet asetting(a, s, d) \in table_items d$
$\forall a : v_arrow; n : NAME; list : \mathbb{P}_1 NAME \bullet aname(a, n, list) \in list$

| $nmax : \mathbb{N}$

$SELECTION == \{i : \mathbb{N} \mid i \leq nmax\}$

| $default_selection : SELECTION$

$MIN == VALUE; MAX == VALUE$

$setting_info_name : ITEM \rightarrow CAPTION$
$setting_value : selection \rightarrow \text{iseq}_l CAPTION$
$setting_info : ITEM \times MIN \times MAX \rightarrow CAPTION$

$\forall s : selection \bullet \text{dom}(setting_value s) = valid s$
--

[$STRING$]

$empty : STRING$
$CHAR : \mathbb{P} INPUT$
$terminator : \mathbb{P} INPUT$
$sprintf : VALUE \rightarrow STRING$

$modify : (STRING \times CHAR) \rightarrow STRING$
--

C.6.1 Therapy console operations

From section 8.3.

```

log_msg : STRING → MESSAGE
o_msg, lo_msg : OPERATOR → MESSAGE
selected_msg, store_msg : NAME → MESSAGE

cancel_run_query : CAPTION
override_query : CAPTION → CAPTION
type_message_prompt, store_field_prompt : CAPTION
delivered_prompt : NAME × VALUE × VALUE × VALUE → CAPTION
exceeded_prompt : NAME × ACCUMULATION × ACCUMULATION → CAPTION

USERNAME == STRING; PASSWORD == STRING

| soper : (USERNAME × PASSWORD) → OPERATOR

list = {select_patient, select_field}
table = {gantry_psa, filter_wedge, leaf_collim, dose_intlk, dose_cal}
setting_table = {gantry_psa, filter_wedge, leaf_collim}
dose_reg_table = {dose_intlk, dose_cal}
table_items = {gantry_psa ↪ {gantry, collim, turnt}, filter_wedge ↪ {filter, wedge, w_rot},
              leaf_collim ↪ leaves, dose_intlk ↪ {p_dose, p_time},
              dose_cal ↪ {pt_mode, press, temp, d_rate, t_fac} }

simple_display == {field_summary, help}
auto_setup_display == {field_summary, filter_wedge, leaf_collim, dose_intlk}
override_table == {filter_wedge, leaf_collim, gantry_psa, dose_intlk}
cal_table == {dose_cal}
dose_table == {dose_intlk}
preset_table == {filter_wedge, leaf_collim}

```

D States and invariants

D.1 Session

From section 5.

Session Vars

mode : MODE
operator : OPERATOR
patient : PATIENT
field : FIELD
names : \mathbb{P} PATIENT
fields : FIELD \rightarrow PRESCRIPTION
counters : FIELD \rightarrow ACCUMULATION

operator = no_operator \vee operator \in operators
mode = experiment \Rightarrow operator \in physicists
names = if mode = therapy then patients else studies

NoPatient

Session Vars

patient = no_patient
field = no_field
fields = \emptyset
counters = \emptyset

PrescribedPatient

Session Vars

patient \neq no_patient
patient \in names
field = no_field \vee field \in dom fields
fields = if mode = therapy then Prescribed patient else Preset patient
mode = therapy \Rightarrow counters = Accumulated patient

$$\text{Session} \cong \text{PrescribedPatient} \vee \text{NoPatient}$$

InitSession _____
 NoPatient

mode = *therapy*
operator = *no-operator*

D.2 Field

From section 6.

Field _____
prescribed : PRESCRIPTION
accumulated : ACCUMULATION
measured : sensor → VALUE
overridden : prescr → VALUE
computed, calibrated : dose-reg → VALUE

PrescribedField _____
Field
Session

field ≠ *no_field*
mode = *therapy* ⇒ *prescribed* = *fields field*

no_prescrip == $(\lambda p : \text{prescrip} \bullet \text{blank})$
no_counter == $(\lambda c : \text{counter} \bullet \text{blank})$
no_dose_reg == $(\lambda d : \text{dose_reg} \bullet \text{blank})$
no_dose == $\{p_dose \mapsto \text{blank}, p_time \mapsto \text{blank}\}$

NoFieldF _____
Field

prescribed = *no_prescrip*
accumulated = *no_counter*
no_dose ⊆ *computed*
overridden = \emptyset

$NoFieldS \hat{=} [Session \mid field = no_field]$
 $NoField \hat{=} NoFieldF \wedge NoFieldS$
 $FieldSession \hat{=} PrescribedField \vee NoField$

$InitField$ _____ $NoFieldF$
$computed = calibrated = no_dose_reg \oplus cal_factor$

E User interface

From section 8.

$Console$ _____ $keyswitch : KEYSWITCH$ $run : RUN$ $display : DISPLAY$ $op : OP$ $interaction : INTERACTION$ $item : ITEM$ $nlist : \mathbb{P} NAME$ $list_item : NAME$ $menu_item : SELECTION$ $buffer : STRING$
--

$InitConsole$ _____ $Console$ _____ $op = login$ $display = login$ $interaction = dialog$ $buffer = empty$

$ConsoleSession$ _____ $Console$ $Session$ $display = select_patient \Rightarrow nlist = names$ $display = select_field \Rightarrow nlist = dom fields$

Console1 _____

Console

new-field : FIELD

new-field ∈ dom fields

Console2 _____

Console

username : STRING

F Reference material

Field	<i>field</i>
Fractions	<i>prescribed n</i>
To date	<i>accumulated n</i>
MU	<i>prescribed dose</i>
Total	<i>prescribed dose_tot</i>
Expected	<i>accumulated n * prescribed dose</i>
To date	<i>accumulated dose_tot</i>

Table 8: Settings and values in the field list display

SETTING	PRESCR	PRESET	ACCUM
DOSE A	<i>prescribed dose</i>	<i>computed p_dose</i>	<i>measured dose</i>
DOSE B	<i>prescribed dose</i>	<i>computed p_dose</i>	<i>measured doseB</i>
TIME	<i>calibrated p_time</i>	<i>computed p_time</i>	<i>calibrated e_time</i>

Table 9: Settings and values in the dosimetry display

	MEASURED/CALIBRATED	ADJUSTED
P/T MODE	<i>computed pt_mode (automatic/manual)</i>	
PRESSURE	<i>calibrated press</i>	<i>computed press</i>
TEMPERATURE	<i>calibrated temp</i>	<i>computed temp</i>
P/T CORR.	<i>calibrated pt_factor</i>	<i>computed pt_factor</i>
CALVOLT 1	<i>calibrated calvolt1</i>	<i>computed calvolt1</i>
CALVOLT 2	<i>calibrated calvolt2</i>	<i>computed calvolt2</i>
DOSE RATE	<i>calibrated d_rate</i>	<i>computed d_rate</i>
TIME FACTOR	<i>calibrated t_fac</i>	<i>computed t_fac</i>

Table 10: Dosimetry calibration display

State precondition	Input precondition	Operation
0 Z_True	RefreshKey	Refresh
0 Z_True	ShutdownKey	Shutdown
0 Unlocked	NoKey	NoOp
1 Available	DisplayKey	SelectDisplay
1 .	PatientKey	SelectPatientList
1 .	TableKey	SelectTable
1 .	MessageKey	TypeMessage
2 List	VArrowKey	GetListArrow
2 Table	ArrowKey	GetSettingArrow
2 PatientSelected	FieldKey	SelectFieldList
2 Setup	LoginKey	Logout
3 Physicist	ExptModeKey	ExptModeC
3 PatientList	SelectKey	SelectPatientC
3 FieldList	SelectKey	SelectFieldC
3 PatientSelected	StoreFieldKey	EditField
3 CalTable	NoKey	NoOp
4 MenuItem	SelectKey	SelectMenuItem
4 DialogItem	SelectKey	SelectDialogItem
3 FieldSelected	NoKey	NoOp
4 AutoSetupDisplay	AutoSetupKey	AutoSetupC
4 OverrideTable	OverrideKey	SelectOverride
4 SettingTable	NoKey	NoOp
5 MenuItem	SelectKey	SelectMenuItem
5 DialogItem	SelectKey	SelectDialogItem
2 Running	CancelRunKey	SelectCancelRun
1 Engaged	NoKey	NoOp
2 LoggedIn	CancelKey	CancelOp
2 Confirm	NoKey	NoOp
4 OverrideOp	SelectKey	OverrideC
4 CancelRunOp	SelectKey	CancelRunC
2 Menu	VArrowKey	GetMenuArrow
3 Editing	SelectKey	MenuSettingC
2 Dialog	CharKey	GetChar
3 LogMessageOp	TerminatorKey	WriteMessageC
3 SelectFieldOp	TerminatorKey	ComplexOrInvalidField
3 StoreFieldOp	TerminatorKey	StoreFieldC
3 Editing	TerminatorKey	EditOrInvalidSetting
3 LoggedOut	CancelKey	CancelUsername
3 UserNameOp	TerminatorKey	EnterUsername
3 PasswordOp	TerminatorKey	LoginOrUnauthorized

Table 11: Therapy console state transition table